
第一章  实验设备介绍

一、 实验设备的联接

[image: image1.png]



图1-1  设备联接

1、 单片机仿真开发器通信电缆接PC机串行口1或2；

2、 单片机仿真开发器仿真头接实验仪；

3、 根据实验需要，实验仪接或不接测量控制对象；

4、 先接通仿真器电源，再接通实验仪电源；

5、 实验过程中，需要连接硬件模块的，先关闭电源再接线，检查正确后打开电源。拆除接线前先关闭电源。

二、 伟福（WAVE）G6W型单片机仿真开发器使用简介
1、 G6W型单片机仿真开发器为MCS—51型仿真开发器；

2、 可选择不同配置的仿真头仿真不同的CPU（如8031、89C51等）；

3、 仿真头上有一个晶振跳线器（选用内、外晶振），出厂设置选用内晶振；

4、 光盘安装编译器软件；

5、 进入编译器时自动进入“仿真器设置”菜单（检查仿真器设置、仿真器状态及联络通信）；

也可在编译器菜单里人工进入“仿真器设置”菜单（实验过程中如果关闭过仿真器、实验仪电源的需要人工进入“仿真器设置”菜单联络通信）；
6、 仿真器设置

1） 伟福汇编器；

2） G6W型仿真器；

3） POD-51型仿真头；

4） 8031 CPU；

5） 不使用伟福软件模拟器；

6） 12M晶振；

7） 程序空间在仿真器上、数据空间在用户板上；

8） PC机与仿真器连接的串行端口为1或2。

7、 仿真器设置失败原因

1） 设置内容错误；

2） PC机串行口故障；

3） PC机与仿真器、仿真器与仿真头连接故障；

4） 仿真器与实验仪未打开电源；

5） 仿真器与实验仪打开电源顺序有误；

6） 仿真器或仿真头损坏；

7） 查无原因的关闭电源重新进行。

8、 编辑、编译、执行及修改文件

1） 编辑及修改

    新建一个文件或打开一个文件，对此文件进行编辑及修改，最后将其保存。汇编文件的文件名后缀为 .ASM。

2） 编译

对当前窗口文件进行编译、校验，并将程序（机器码）传至仿真器；

如果程序文件有错，信息窗口将提示出错信息，以便用户修改。

3） 执行与退出

包括全速、跟踪、单步、执行到光标处等几种程序执行方式；

执行前可设置断点；

退出有暂停和复位。

9、 在“窗口”中检查RAM及特殊功能寄存器的内容

全速运行程序后，要先“暂停”再在“窗口”中检查存储器内容；

CPU窗口包含源程序的反汇编代码、特殊功能寄存器状态等；

数据窗口包含内部RAM和外部RAM的状态等。

三、 综合实验仪介绍
1、 电源部分

外接220V交流电源；

实验板上有以下电源插孔：+2.5V, +5V, +12V, -5V, -12V及0V（GND）；

实验板上每一硬件模块的电源独立控制，由短路块连通。
2、 测量电桥（见下页图1-2）

电源+2.5V；

Rx为热敏电阻Pt100（调试时接电阻箱）；

输出Vout11、Vout12；

WR3为调零电位器。

3、 测量放大器（见下页图1-3）
电源+12V，-12V；

输入Vin41、Vin42, 输出Vout4；

WR5为增益调整电位器，WR6为调零电位器。

4、 电压衰减器（见下页图1-4）

电源+12V，-12V；

输入Vin2, 输出Vout2；

WR4调节衰减系数。 

图1-2  测量电桥

图1-3  测量放大器

图1-4  电压衰减器
5、 D/A转换电路（DAC0832）

电源+5V，+12V，-12V；

输入D7~D0，输出Vout3；

写信号/WR，片选信号/CS，标准电压端VREF，反馈电阻端RFB，反馈端FG。
[image: image4.png]



图1-5   D/A转换电路（DAC0832）
6、 A/D转换电路（ADC0809）

    电源+5V；
输入IN0~IN7，输出D0~D7；

读信号/RD，写信号/WR，时钟信号CLK，转换结束信号EOC，片选信号/CS；

地址A、B、C已接D0、D1、D2。

图1-6
7、 四位半A/D转换电路(ICL7135)
电源+2.5V, +5V, -5V。

输入Vin5。使用单端输入INH1，INL0（负端输入）已接地。输入范围为0V~1.9999V，对应的输出数据为00000~19999（BCD码）。

转换状态信号BUSY。A/D转换时为高电平，转换结束时为低电平，可供查询或中断申请用。

时钟CLKIN。接1/8 ALE，单片机晶振频率为12MHZ，则1/8 ALE为250KHZ，转换频率约为6次/秒。转换控制端R/H已接高电平，表示ICL7135将连续自动转换。

过量程标志OVER。当输入信号超过1.9999V时，输出高电平。

欠量程标志UNDER。当输入信号小于量程的9%（0.1800V）时，输出高电平。但在单端输入时，输入信号在0V~0.1800V时并不影响A/D转换器的正常工作。

位状态输出D5、D4、D3、D2、D1。转换结束以后的任一时刻，只有一位输出高电平，其它输出低电平。分别表示现时数据输出为万、千、百、十、个位。

数据输出B8、B4、B2、B1。以BCD码形式输出（0000~1001），分时输出万、千、百、十、个位数据（00000~19999）。

WR7调整ICL7135第2脚（标准电压端VREF）电压为1.0000V。
[image: image6.png]



图1-7   A/D转换电路(ICL7135)
8、 8031扩展I/O口8255（1）

电源+5V；

8031引出端 /INT0、/INT1、T0、T1、/WR、/RD、P0.0~P0.7；
8255（1）片选信号为 /CS，PA口为PA7~PA0，PB口为PB7~PB0，PC口为PC7~PC0；
译码电路74138输出 /Y3 ~ /Y7，地址范围为6000H~FFFFH；
双分频电路74393输出为1/2ALE、1/4 ALE、1/8 ALE（输入已接ALE）。
原理图见下页图1-8。
[image: image7.png]



图1-8  8031扩展8255（1）原理图

9、 利用8255（2）扩展LED显示器及行列式键盘

电源+5V；

8255（2）片选信号为 /CS；

LED显示器为共阴极，MC1413为反向驱动器，三极管9013为正向驱动。

原理图见下页图1-9。

图1-9  8255（2）扩展LED显示器及行列式键盘原理图

10、RS-422串行口

电源+5V；

输入为IN（+/-），输出为OUT(+/-)。
[image: image9.png]- oy





图1-10  RS-422串行口
11、固态继电器（负载）控制电路

电源+5V；

控制信号输入为CONin, 输出为CONout(+/-)；

控制信号CONin为高时，负载接通电源。
[image: image10.png]L





图1-11  固态继电器控制电路
12、555脉冲电路

电源+5V；

    输出为Fout；
WR8调节输出信号频率。

见图1-12。                               
[image: image11.png]



图1-12  555脉冲电路
13、拨码盘电路

电源+5V；

输出为高位“8421” 端口和低位“8421” 端口。

[image: image12.png]



图1-13

14、钮子开关电路

电源+5V；

输出为K0~K7；

钮子开关拨向上方时输出为高“1”，拨向下方时输出为低“0”。

[image: image13.png]



图1-14  钮子开关电路

15、LED发光二极管电路

电源+5V；

输入为L0~L7；

输入为高时发光二极管亮，输入为低时发光二极管灭（7406为反向驱动器）。
[image: image14.png]



图1-15   LED发光二极管电路

16、单脉冲输出电路
电源+5V；

K为按钮。

[image: image15.png]



 图1-16  单脉冲输出电路

第二章  基本程序设计实验

实验一  数据传送实验

一、目的：

熟悉单片机仿真开发器的使用；

掌握编程的方法；

掌握8031内部RAM的数据操作。

二、内容： 

内部RAM 40H~4FH单元置数A0H~AFH，然后将40H~AFH 单元中内容送到内部RAM 50H~5FH单元中。

编辑、编译、运行程序，检查内部RAM中的结果。
三、参考程序：
         ORG  0000H

 RESET： AJMP  MAIN

         ORG  003FH

 MAIN：  MOV  R0，#40H

         MOV  R2，#10H

         MOV  A，#0A0H

  A1：   MOV  @R0,A

         INC   R0

         INC   A

         DJNZ  R2,A1

         MOV  R0, #40H

         MOV  R1, #50H
MOV  R2, #10H

  A2:    MOV  A,@R0

         MOV  @R1, A

         INC   R0

         INC   R1

         DJNZ  R2, A2

  A3:    SJMP  A3
四、分析以上程序。

五、运行程序后（暂停），检查并记录相关单元内容。
*六、按以上步骤，验证教材中程序设计的例题。

实验二  多字节十进制加法实验

一、目的：

掌握MCS-51汇编语言程序设计方法。

二、内容：

多字节的十进制加法。

加数存储单元首地址由R0指出，被加数和结果的存储单元首地址由R1指出，字节数由R2指出。

编辑、编译、运行程序，检查内部RAM中的结果。

三、参考程序：

加数存储单元为：31H、30H，被加数存储单元为：21H、20H，结果存储单元为：22H、21H、20H。

ORG  0000H

 RESET:  AJMP MAIN

         ORG  0100H

  MAIN:  MOV  SP, #60H

         MOV  R0, #31H

         MOV  @R0, #22
         DEC  R0

         MOV  @R0, #33
         MOV  R1, #21H

         MOV  @R1, #44
         DEC  R1

         MOV  @R1, #55
         MOV  R2, #02H

         ACALL  DACE

  HERE:  SJMP  HERE

  DACE:  CLR   C

  DAL:   MOV  A,@R0

         ADDC  A,@R1

         DA   A

         MOV   @R1,A

         INC    R0

         INC    R1

         DJNZ  R2,DAL

         CLR   A

         MOV  ACC.0,C
         MOV   @R1,A
         RET
四、分析以上程序。

五、运行程序后（暂停），检查并记录相关单元内容。

六、修改加数和被加数（十进制数），重复以上过程。

七、填写下列单元地址及内容：

加数单元地址      （   ）（   ）                   

加数单元地址      （   ）（   ）

结果单元地址（   ）（   ）（   ）

*八、按以上步骤，验证作业中程序设计的习题。
第三章  I/O口（8255）扩展及外部中断实验

实验三  8255输入输出实验

一、目的：

掌握8031单片机并行口扩展方法；熟悉8255芯片性能，掌握其编程方法。

二、内容：

8255（1）从PA口输入，从PB口输出。
硬件连接框图见图3-1。

相关电路原理见第一章图1-8、图1-14和图1-15。

图3-1
三、实验步骤：

1、 硬件连接：

钮子开关K0~K7连接8255（1）的PA0~PA7；
    发光二极管L0~ L7连接8255（1）的PB0~PB7；

    8255（1）片选信号 /CS连接 /Y3 ~/Y7（任选其一）；

     将相关硬件模块的电源短路块接通。

2、 软件设计：

8255（1）初始化，数据从PA口输入，取反后从PB口输出。

3、 执行程序，观察并记录实验结果。

四、参考程序框图见图3-2。

*五、查询资料，了解其它常用的并行I/O口芯片。

图3-2  参考程序框图

六、参考程序：

    参考程序中8255（1）片选信号 /CS连接 /Y3。

ORG 0000H

AJMP MAIN

        ORG 0100H

MAIN:  MOV DPTR , #6003H

        MOV A , #90H

        MOVX @DPTR , A

L1:     MOV DPTR , #6000H

        MOVX A , @DPTR

        CPL A

        MOV DPTR , #6001H

        MOVX @DPTR , A

        SJMP L1
实验四  外部中断实验

一、目的：

了解单片机中断原理、中断过程、中断方式的选择及编程方法。

二、内容：

利用单脉冲输出电路的脉冲，通过 /INT0向CPU申请外部中断，CPU响应中断后，执行中断服务子程序。

    中断服务子程序内容为：从8255（1）PA口输入数据，取反后从PB口输出。

    硬件连接框图见图3-3。

    相关电路原理图见第一章图1-8、图1-14、图1-15和图1-16。

图3-3

三、实验步骤：

1、 硬件连接： 

钮子开关K0~K7连接PA0~PA7；

发光二极管L0~ L7连接 PB0 ~ PB7；

8255（1）片选信号 /CS连接 /Y3 ~ /Y7（任选其一）；
    单脉冲输出电路负脉冲输出端连接 /INT0；

    将相关硬件模块的电源短路块接通。

2、 软件设计：

主程序中等待外部中断；

    中断服务子程序中数据从PA口输入，取反后从PB口输出。 

3、 执行程序，观察并记录实验结果。

四、参考程序框图见图3-4。

*五、8051单片机5个中断源的中断矢量入口地址？哪些中断申请标志由CPU自动清除？哪些中断申请标志必须由软件清除？

图3-4  参考程序框图
六、参考程序：
参考程序中8255（1）片选信号 /CS连接 /Y3。

org 0000h

                ajmp main

                org 0003h

                ajmp int
                org 0100h

        main:   mov dptr , #6003h

                mov a , #90h

                movx @dptr , a         ;8255PA口输入PB口输出

                setb ea

                setb ex0

                setb it0                ;/INT0边沿触发

                nop

                nop

                sjmp $

         int:    mov dptr , #6000h

                movx a , @dptr

                cpl a

                mov dptr , #6001h

                movx @dptr , a

                reti
第四章  显示器、键盘及BCD拨码盘扩展实验

实验五  动态LED显示实验

一、目的：

了解动态LED显示器电路的结构。

通过实验掌握动态LED显示器软硬件设计方法。
二、内容：

设计程序在六位LED显示器上显示字符“1 2 3 4 5 6”。

硬件连接框图见图4-1。

    LED显示器电路原理图见第一章图1-9。

图  4-1
三、实验步骤：

1、 硬件连接：

    8255（2）片选信号 /CS连接 /Y3 ~ /Y7（任选其一）；

将相关硬件模块的电源短路块接通。

2、 软件设计：

1） 8255（2）PA口PA7控制小数点（正向驱动），PA6~PA0控制

字形（反向驱动）； 
8255（2）PB口PB5~PB0控制字位（反向驱动）；

LED显示器为共阴极。

2） 2ms软件延时子程序：
                  DELAY2MS:     MOV R3, #04H

                       DELAY0:     MOV R4, #0FFH

                       DELAY1:     DJNZ R4, DELAY1

                                   DJNZ R3, DELAY0

                                   RET
3） 字形码（0~F）：

                       DB  40H,79H,24H,30H  

                       DB  19H,12H,02H,78H

                       DB  00H,10H,08H,03H

                       DB  46H,21H,0EH,41H

3、 执行程序，观察并记录实验结果。

四、参考程序框图见下页图4-2。
*五、高级要求：

1、 显示小数点。在左边第3位后面增加小数点的显示，该位字形码如何确定？修改程序，观察结果。

2、 设计程序，使显示器产生动感显示效果，显示字符从左向右移动。

图4-2  程序框图

六、参考程序：

参考程序中8255(2) 片选信号/CS接 /Y3。




          org 0000h

                    ajmp main

                     org 0100h

            main:   mov dptr,#6003h     ;8255(2)初始化

                    mov a,#80h          ;PA口PB口输出

                    movx @dptr,a        ;

           loop2:   mov r7,#06h         ;LED显示六位

                    mov r6,#01h         ;从左边第一位开始显示

                    mov r5,#01h         ;从“1”开始显示

           loop1:   mov a,r6            ;PB口送位码

                    mov dptr,#6001h      ;

                    movx @dptr,a        ;

                    mov a,r5            ;查字形码

                    mov dptr,#dsp        ;

                    movc a,@a+dptr      ;

                    mov dptr,#6000h      ;送字形码

                    movx @dptr,a        ;

                    acall delay2ms       ;延时2MS

                    mov a,r6            ;指向下一位

                    rl a                 ;

                    mov r6,a             ;

                    inc r5               ;显示下一个数

                    djnz r7,loop1         ;六位显示完？

                    ajmp loop2          ;六位显示完，重新开始

         delay2ms:  mov r3,#04h         ;2MS延时子程序

         delay0:    mov r4,#0ffh         ;

         delay1:    djnz r4,delay1        ;

                    djnz r3,delay0        ;

                    ret                 ;

         dsp:       db 40h,79h,24h,30h   ;字形表

                    db 19h,12h,02h,78h

                    db 00h,10h,08h,03h

                    db 46h,21h,06h,0eh 

实验六  BCD拨码盘扩展实验

一、 目的：

掌握BCD拨码盘的扩展方法。

二、内容：

读入两位BCD拨码盘的数据，在LED显示器上显示。

硬件连接框图见图4-3。
相关原理图见第一章图1-8、图1-9及图1-12。

图 4-3

三、实验步骤：
1、 硬件连接：

将BCD拨码盘高位、低位“8421”输出分别接8255（1）PA口的PA7 ~PA4、

PA3 ~PA0；

    8255（1）片选信号 /CS连接 /Y3 ~ /Y7（任选其一）；

    8255（2）片选信号 /CS连接 /Y3 ~ /Y7（任选其一）；
    将相关硬件模块的电源短路块接通。
2、 软件设计：

1） 8255（1）PA口输入；

2） 8255（2）PA口PA7控制小数点, 8255（2）PA口PA6~PA0控制字形（ 

反向驱动）；

8255（2）PB口PB5~PB0控制字位（反向驱动）；

LED显示器为共阴极。 

*四、高级要求：

将两位BCD数据读入后相乘，积送去显示。（先把BCD码变换成16进制数，相乘后再变换成BCD码，送去显示。）

五、参考程序：

参考程序中8255（1）片选信号 /CS连接 /Y7；8255（2）片选信号 /CS连接 /Y3。

ORG 0000H

          AJMP START

          ORG 0100H

   START: MOV R0 , #50H

          MOV R2 , #06H

   LOOP0: MOV @R0 , #10H          ;给显示缓冲区50H-55H赋值

          INC R0                     ;显示全黑

          DJNZ R2,LOOP0

          MOV DPTR,#6003H

          MOV A,#80H

          MOVX @DPTR,A    ;对8255(2)进行初始化(接LED显示器)

          MOV A,#90H        ;对8255(1)进行初始化(接BCD拨盘)

          MOV DPTR,#0E003H

          MOVX @DPTR,A

    MAIN:

          ACALL LED                  ;调显示子程序。

          MOV DPTR,#0E000H           ;读BCD值。

          MOVX A,@DPTR

          MOV B,A

          ANL A,#0F0H

          SWAP A

          MOV 50H,A

          MOV A,B

          ANL A,#0FH

          MOV 51H,A

          AJMP MAIN

     LED:

          MOV R0,#01H                ;位码

          MOV R1,#50H                ;显示缓冲区首址

    LOOP:

          MOV A,R0                   ;送位码

          MOV DPTR,#6001H

          MOVX @DPTR,A

          MOV A,@R1                  ;送段码

          MOV DPTR,#SEG

          MOVC A,@A+DPTR

          MOV DPTR,#6000H

          MOVX @DPTR,A

          ACALL DELAY

          MOV A,#7FH                 ;清黑

          MOVX @DPTR,A

          INC R1

          MOV A,R0                   ;指向下一位

          RL A

          MOV R0,A

          CJNE R0,#04H,LOOP          ;两位显示完返回

          RET

   DELAY: MOV R7,#02H

  DELAY0: MOV R6,#0FFH

  DELAY1: DJNZ R6,DELAY1

          DJNZ R7,DELAY0

          RET

     SEG: DB 40H,79H,24H,30H,19H,12H,02H,78H  ;字形表

          DB 00H,10H,08H,03H,46H,21H,06H,0EH

实验七  键盘显示器扩展实验

1、 目的：
了解键盘显示器扩展电路的结构；

掌握行列式键盘输入程序设计方法。
2、 内容：

从键盘上输入0~9及A~F，在LED显示器上显示。

硬件连接框图见图4-4。

键盘显示器原理图见第一章图1-9。

图 4-4
三、实验步骤：
1、 硬件连接：

    8255（2）片选信号 /CS连接 /Y3 ~ /Y7（任选其一）；
    将相关硬件模块的电源短路块接通。
2、 软件设计：

1） 8255（2）PA口PA7控制小数点, 8255（2）PA口PA6~PA0控制字形（ 

反向驱动）；

8255（2）PB口PB5~PB0控制字位（反向驱动）；

LED显示器为共阴极。 

2） 8255（2）PC口PC7~PC4为行列式键盘的行线；

PC3~PC0 为行列式键盘的列线；

行线、列线均由上拉电阻接+5V。

3） 行线输出全零时读列线输入；列线输出全零时读行线输入。

4） 键值表（0~9及A~F键按下时，PC7~PC4 、PC3~PC0输入合并后的数值）       

                  DB 7EH,0BEH,0BDH,0BBH     

                  DB 0DEH,0DDH,0DBH,0EEH

                  DB 0EDH,0EBH,0E7H,0D7H

DB  0B7H,77H,7BH,7DH
3、 执行程序，观察并记录实验结果。

*四、高级要求：

查询资料，举出其它形式的键盘、显示器软硬件设计方法。分析优缺点。

五、参考程序：

参考程序中8255(2) 片选信号/CS接 /Y3。

        ORG 0000H

        AJMP START

        ORG 0100H

START:  MOV A,#06H      ;六位显示缓冲区（75H-7AH）清零

        MOV R0,#75H     ;

CLE:    MOV @R0,#00H;

        INC R0;

        DEC A;

        JNZ CLE;

KEY:    ACALL LED        ;调显示子程序

        MOV DPTR,#6003H  ;扫描键盘。8255（2）初始化

        MOV A,#88H        ;PA口PB口输出（LED显示器）

        MOVX @DPTR,A    ;PC口高四位输入低四位输出（行列式键盘）
        MOV DPTR,#6002H  ;PC口低四位输出零

        MOV A,#00H        ;

        MOVX @DPTR,A    ;

        MOVX A,@DPTR    ;PC口高四位输入

        ANL A,#0F0H        ;

        MOV 74H,A         ;键值高四位送74H

        CJNE A,#0F0H,READ ;键值高四位非全“1”有键按下

        AJMP KEY          ;键值高四位全“1”无键按下

READ:   MOV DPTR,#6003H   ;有键按下。8255（2）初始化（行列输入输出                   

        MOV A,#81H         ;  交换），PA口PB口输出（LED显示器），

        MOVX @DPTR,A     ;PC口高四位输出低四位输入（行列式键盘）
        MOV DPTR,#6002H   ;PC口高四位输出零

        MOV A,#00H         ;

        MOVX @DPTR,A     ;

        MOVX A,@DPTR     ;PC口低四位输入

        ANL A,#0FH         ;

        ADD A,74H          ;键值低四位送74H

        MOV 74H,A         ;

        ACALL DELAY12MS  ;软件延时

KEY1:   ACALL LED

        MOV DPTR,#6003H  ;再扫描一次

        MOV A,#88H        ;

        MOVX @DPTR,A      ;
        MOV DPTR,#6002H    ;

        MOV A,#00H         ;

        MOVX @DPTR,A     ;

        MOVX A,@DPTR     ;

        ANL A,#0F0H        ;
        MOV 73H,A         ;
        MOV DPTR,#6003H  ;

        MOV A,#81H        ;

        MOVX @DPTR,A    ;
        MOV DPTR,#6002H  ;

        MOV A,#00H        ;

        MOVX @DPTR,A     ;

        MOVX A,@DPTR     ;

        ANL A,#0FH         ;

        ADD A,73H          ;

        CLR C             ;两次值相同去再扫描一次（按键未松）

        SUBB A,74H        ;

        JZ KEY1            ;

        MOV R1,#00H      ; 按键已松。R1作计数器

        MOV R5,#10H      ; 16个键
SEARCH: MOV DPTR,#KTABLE  ;键值表首址

        MOV A,R1             ;

        MOVC A,@A+DPTR    ;查键表值

        CLR C                ;

        SUBB A,74H          ;比较读入键值与键值表数据

        JZ FIND              ;相同则键值有效，去处理显示缓冲区

        INC R1              ;不相同比较下一个键值表数据

        DJNZ R5,SEARCH    ;

        AJMP KEY          ;16个键值表数据比较完

FIND:   MOV A,R1          ;键值有效准备显示

        MOV R5,A          ;R1数值与键盘标称值（0-F）相同

        MOV R7,#05H       ;

        MOV R0,#79H       ;显示缓冲区（7AH-75H）数据向前移位

        MOV R1,#7AH       ;

ROL:    MOV A,@R0        ;

        MOV @R1,A         ;

        DEC R0            ;

        DEC R1            ;

        DJNZ R7,ROL       ;

        MOV 75H,R5       ;新读键盘标称值（0-F）送显示缓冲区75H
        AJMP KEY         ;

LED:    MOV DPTR,#6003h  ;显示子程序

        MOV A,#80h        ;8255（2）初始化

        MOVX @DPTR,A     ;
SHOW:   MOV R0,#75h      ;显示缓冲区首地址

        MOV R2,#01h       ;位码

        MOV A,R2          ;
LOOP:   MOV DPTR,#6001H  ;位码送PB口

        MOVX @DPTR,A     ;
        MOV DPTR,#6000H   ;PA口送全黑字形码

        MOV A,#7FH         ;

        MOVX @DPTR,A     ;

        MOV DPTR,#LTABLE  ;字形表首地址

        MOV A,@R0          ;取显示缓冲区值

        MOVC A,@A+DPTR   ;取字形码
        MOV DPTR,#6000H   ;字形码送PA口

        MOVX @DPTR,A     ;

        ACALL DELAY1MS   ;软件延时
        INC R0              ;指向下一显示缓冲区地址

        MOV A,R2           ;

        JB ACC.5,LOOP1     ;第六位显示完？

        RL A               ;未完显示下一位

        MOV R2,A          ;

        AJMP LOOP        ;

LOOP1:  RET               ;第六位显示完返回

DELAY12MS: MOV R3,#0FFH  ;

DLAY0:     MOV R4,#0FFH

DLAY1:     DJNZ R4,DLAY1

           DJNZ R3,DLAY0

           RET

DELAY1MS:  MOV R3,#02H  ;

DELAY0:    MOV R4,#0FFH

DELAY1:    DJNZ R4,DELAY1

           DJNZ R3,DELAY0

           RET

LTABLE: DB 40H,79H,24H,30H,19H    ;字形表

        DB 12H,02H,78H,00H,10H

        DB 08H,03H,46H,21H,06H

        DB 0EH

KTABLE: DB 7EH,0BEH,0BDH,0BBH  ;键值表

        DB 0DEH,0DDH,0DBH,0EEH

        DB 0EDH,0EBH,0E7H,0D7H

        DB 0B7H,77H,7BH,7DH

        END

第五章  计数器定时器实验

实验八  计数器实验

一、目的：

掌握计数器的使用方法，学会计数器工作方式的设置。

二、内容：

利用计数器对单脉冲电路产生的脉冲进行计数，并在LED显示器上显示。硬件连接框图见图5-1。
相关电路原理图见第一章图1-9及图1-16。

图 5-1
三、实验步骤：
1、 硬件连接：

将单脉冲电路的正脉冲输出端连接8031计数器T0的输入端T0；

8255（2） 片选信号 /CS连接 /Y3 ~ /Y7（任选其一）；

    将相关硬件模块的电源短路块接通。

2、 软件设计：

定时器/计数器T0设置成计数方式，工作方式为方式1，计数初值0000H，

软件启动；    

对单脉冲电路的正脉冲进行计数，并将计数值在LED显示器上显示；

8255 （2）PA口PA7控制小数点，PA6~PA0控制字形（反向驱动）；

8255（2）PB口PB5~PB0控制字位（反向驱动）；

LED显示器为共阴极。

3、 执行程序，观察并记录实验结果。

*四、高级要求：

要求将计数器中的数值以十进制数的形式显示。修改程序，观察结果。

五、参考程序：
参考程序中8255(2) 片选信号/CS接 /Y3。

       ORG 0000H

       AJMP START

       ORG 0100H

START: MOV TMOD,#05H;      T0初始化。计数方式1，软件启动

       MOV TH0,#00H;        初值

       MOV TL0,#00H;

       CLR EA;               关中断

       CLR ET0;

       SETB TR0;             启动计数器

       MOV DPTR,#6003H;     8255初始化（扩展LED显示器）

       MOV A,#80H

       MOVX @DPTR,A

       MOV R7,#06H;          显示缓冲区清0

       MOV R1,#70H;

CLE:   MOV @R1,#00H;

       INC R1

       DJNZ R7,CLE

READT0: ACALL LED;

        MOV A,TH0;         读计数器

        MOV 41H,A;

        MOV A,TL0;

        MOV 40H,A;

        MOV A,41H;

        CJNE A,TH0,READT0; 处理进位

        MOV 73H,A;         高位拆字

        ANL A,#0FH

        MOV 72H,A

        MOV A,73H

        ANL A,#0F0H

        SWAP A

        MOV 73H,A

        MOV A,40H;         低位拆字

        MOV 71H,A

        ANL A,#0FH

        MOV 70H,A

        MOV A,71H

        ANL A,#0F0H

        SWAP A

        MOV 71H,A

        AJMP READT0

LED:    MOV R7,#06H;       显示子程序

        MOV R0,#75H;       显示缓冲区75H-70H

        MOV R2,#01h;       位码

        MOV A,R2

LOOP:   MOV DPTR,#6001H;    B口位选

        MOVX @DPTR,A

        MOV DPTR,#LTABLE;   查字形码

        MOV A,@R0

        MOVC A,@A+DPTR

        MOV DPTR,#6000H;    A口字选

        MOVX @DPTR,A

        ACALL DELAY2MS

        DEC R0;

        MOV A,R2

        RL A

        MOV R2,A

        DJNZ R7,LOOP;

        RET

DELAY2MS: MOV R3,#04H

DELAY0:   MOV R4,#0FFH

DELAY1:   DJNZ R4,DELAY1

          DJNZ R3,DELAY0

          RET

LTABLE: DB 40H,79H,24H,30H

        DB 19H,12H,02H,78H

        DB 00H,10H,08H,03H

        DB 46H,21H,06H,0EH

实验九  定时器（时钟）实验

1、 目的：

掌握定时器的使用方法，学会定时器工作方式的设置。

2、 内容：

在LED显示器上显示时、分、秒。

硬件连接框图见图5-2。

LED显示器电路原理图见第一章图1-9。

图 5-2
三、实验步骤：

1、 硬件连接：

    将8255（2）片选信号 /CS连接 /Y3 ~ /Y7（任选其一）；

    将相关硬件模块的电源短路块接通。

2、 软件设计：

定时器/计数器T1设置成定时方式，工作方式为方式1，软件启动，允

许中断（晶振频率为12M）；

利用定时器T1产生50MS定时中断，20次中断后秒加1，60秒后分加1，

60分后时加1，24时后恢复0时；

将时、分、秒送入LED显示器显示；

8255（2）PA口PA7控制小数点，PA6~PA0控制字形（反向驱动）；

PB 口PB5~PB0控制字位（反向驱动）；

LED显示器为共阴极。

3、 执行程序，观察并记录实验结果。

*四、高级要求：

1、 利用键盘设置时钟初值，并利用键盘控制时钟的暂停和启动。修改程序，

观察结果。

2、 查询资料，了解专用时钟芯片及其与8031单片机的连接方式。
五、参考程序：

参考程序中8255(2) 片选信号/CS接 /Y3。
           ORG  0000H

            AJMP  MAIN        ; 

             ORG  001BH        ; T1的中断入口地址

            AJMP  SERVE        ; 转向中断服务程序

             ORG  0100H

    MAIN:   MOV  TMOD,#10H    ; 设T1工作于定时模式1

            MOV  TH1,#3CH

            MOV  TL1,#0B0H     ; 赋计数初值（12MHZ时50MS）

            MOV  DPTR,#6003H   ; 8255(2)初始化

            MOV  A,#80H

            MOVX  @DPTR,A

            MOV  20H,#14H       ; 装入中断次数（20次）

            CLR  A

            MOV  30H,A           ; 时单元清0（时分秒初值）

            MOV  31H,A           ; 分单元清0

            MOV  32H,A           ; 秒单元清0

            SETB  ET1              ; 允许T1中断

            SETB  EA               ; 允许CPU中断

            SETB  TR1              ; 启动定时器T1

  LOOP:     ACALL  LED            ; 调用显示子程序（等待中断）

            AJMP  LOOP

   LED:     MOV  A,30H            ; 显示子程序

            ANL  30H,#0F0H         ；将时单元拆为40与41单元

            MOV  40H,30H

            MOV  30H,A

            ANL  30H,#0FH

            MOV  41H,30H

            MOV  A,31H            ; 将分单元拆为42与43单元

            ANL  31H,#0F0H

            MOV  42H,31H

            MOV  31H,A

            ANL  31H,#0FH

            MOV  43H,31H

            MOV  A,32H            ; 将秒单元拆为44与45单元

            ANL  32H,#0F0H

            MOV  44H,32H

            MOV  32H,A

            ANL  32H,#0FH

            MOV  45H,32H

            MOV  R0,#40H        ; R0指向显示缓冲区首位

            MOV  R1,#01H        ; 送位码到R1

            MOV  R5,#06H        ; R5作计数器，显示6位 

  LOOP1:    MOV  DPTR,#6001H  ; 8255的PB口送位选码

            MOV  A,R1

            MOVX  @DPTR,A

            MOV  DPTR,#SEG

            MOV  A,@R0

            MOVC  A,@A+DPTR 

            MOV  DPTR,#6000H      ; 8255的PA口送段选码

            MOVX  @DPTR,A

            ACALL  DELAY1MS       ; 调用延时子程序

            INC  R0                 ; 指向下一个缓冲区

            MOV  A,R1

            RL  A                   ; 指向下一个显示位

            MOV  R1,A

            DJNZ  R5,LOOP1

            RET

  DELAY1MS: MOV  R7,#02H            ; 延时1毫秒

    DELAY0: MOV  R6,#0FFH

    DELAY1: DJNZ  R6,DELAY1

            DJNZ  R7,DELAY0

            RET

SEG:    DB  40H,79H,24H,30H,19H

            DB  12H,02H,78H,00H,10H    ; 0-9字形表

   SERVE:   PUSH  PSW              ;  中断服务子程序

            PUSH  ACC              ;  保护现场

            MOV  TH1,#3CH

            MOV  TL1,#0B0H       ; 重新赋计数初值

            DJNZ  20H,RETUNT      ; 1秒未到，返回

            MOV  20H,#14H        ; 重置中断次数

            MOV  A,#01H

            ADD  A,32H            ; “秒位”加1

            DA   A

            MOV  32H,A           ; 转换为BCD码

            CJNE  A,#60H,RETUNT   ; 未计满60秒，返回

            MOV  32H,#00H        ; 计满60秒，“秒位”清0

            MOV  A,#01H

            ADD  A,31H            ;“分位”加1

            DA   A

            MOV  31H,A           ; 转换为BCD码

            CJNE  A,#60H,RETUNT  ; 未计满60分，返回

            MOV  31H,#00H        ; 计满60分，“分位”清0

            MOV  A,#01H

            ADD  A,30H            ; “时位”加1

            DA   A

            MOV  30H,A            ; 转换为BCD码

            CJNE  A,#24H,RETUNT     ; 未计满24小时，返回

            MOV  30H,#00H          ; 计满24小时，“时位”清0

   RETUNT:  POP  ACC

            POP  PSW               ; 恢复现场

            RETI                     ; 返回

第六章  串行通信实验
实验十  串行口实验

一、目的：

利用串行口实现单片机间的通信。

二、内容：

将A单片机的内存单元40H~4FH的内容通过串行口传输到B单片机中，存放在B单片机的内存单元50H~5FH。
硬件连接框图见图6-1。

    串行口电路原理图见第一章图1-10。

图6-1

三、实验步骤：

1、 硬件连接：

    A单片机实验仪RS-422接口输入端IN(+,-)连接B单片机实验仪RS-422接口输出端OUT(+,-)；

    B单片机实验仪RS-422接口输入端IN(+,-)连接A单片机实验仪RS-422接口输出端OUT(+,-)。
连接相关硬件模块的电源短路块。
2、 软件设计：

1） A、B单片机串行口工作方式为方式1，定时器T1做波特率发生器，波特率为1200HZ（晶振频率为12M）；

2） A机在程序中给40H~4FH单元赋值；

3） B机接收的数据存放在50H~5FH单元中。

3、 执行程序后，暂停。从窗口中检查B机50H~5FH单元中的内容。

四、参考程序框图见下页图6-2。

*五、高级要求：

要求先将A单片机的内存单元40H~4FH的内容通过串行口传输到B单片机中，存放在B单片机的内存单元50H~5FH；

然后再将B单片机的内存单元40H~4FH的内容通过串行口传输到A单片机中，存放在A单片机的内存单元50H~5FH。

A、B单片机之间利用联络字进行协调。

图6-2   串行口实验程序流程图（高级要求）
六、参考程序：
1、 发送：

                ORG   0000H

                AJMP  MAIN

                ORG   0100H

   MAIN:        MOV   R0,#40H       ;40H~4FH赋值

                MOV   R7,#10H

                MOV   A,#00H

   LOOP:        MOV   @R0,A

                INC   A

                INC   R0

                DJNZ  R7,LOOP

   FMT_T_S:     MOV   TMOD,#20H
     ;波特率设置



         MOV   TH1,#0F3H



         MOV   TL1,#0F3H



         SETB  TR1



         MOV   SCON,#50H      ;串行口初始化



         MOV   PCON,#80H      ;置SMOD＝1
   FMT_RAM:     MOV   R1,#40H        ;数据首址送R1

                MOV   R6,#00H        ;清累加和寄存器

                MOV   R5,#10H        ;送字节数至R5

   TX_ACK:      MOV   A,#06H         ;发呼叫信号“06H”

                MOV   SBUF,A

   WAIT1:       JBC   TI,RX_YES      ;等待发送完

                SJMP  WAIT1

   RX_YES:      JBC   RI,NEXT1       ;接收乙机回答

                SJMP  RX_YES

   NEXT1:       MOV   A,SBUF         ;判断乙机是否同意接收



         CJNE  A,#00H,TX_ACK

   TX_BYTES:    MOV   A,R5           ;向乙机发送要传送的字节个数

                MOV   SBUF,A

   WAIT2:       JBC   TI,TX_NEWS



         SJMP  WAIT2

   TX_NEWS:  
  MOV   A,@R1          ;发送

                MOV   SBUF,A

                ADD   A,R6



         MOV   R6,A

                INC   R1             ;指向下一个数据

   WAIT3:       JBC   TI,NEXT2

                SJMP  WAIT3

   NEXT2:       DJNZ  R5,TX_NEWS     ;判发送结束

   TX_SUM:      MOV   A,R6           ;数据发送完，发累加和给乙机

                MOV   SBUF,A

   WAIT4:       JBC   TI,RX_0FH

                SJMP  WAIT4

   RX_0FH:      JBC   RI,IF_0FH        ;等待乙机回答

                SJMP  RX_0FH

   IF_0FH:      MOV   A,SBUF

                CJNE  A,#0FH,FMT_T_S   ;判断传送正确否

                SJMP  $

2、 接收：

                ORG   0000H

                AJMP  MAIN

                ORG   0100H

   MAIN:        MOV   TMOD,#20H      ;T1初始化



         MOV   TH1,#0F3H

                MOV   TL1,#0F3H



         SETB  TR1



         MOV   SCON,#50H
     ;串行口初始化



         MOV   PCON,#80H      ;SMOD=1

   FMT_RAM:     MOV   R1,#50H         ;数据首址送R1

                MOV   R6,#00H         ;清累加和寄存器

   RX_ACK:
     JBC   RI,IF_06H        ;接收呼叫信号



         SJMP  RX_ACK

   IF_06H:      MOV   A,SBUF         ;判断呼叫信号是否有误

                CJNE  A,#06H,TX_15H

   TX_00H:      MOV   A,#00H         ;向甲机回送同意接收信号

                MOV   SBUF,A

   WAIT1:       JBC   TI,RX_BYTES    ;等待应答信号发送完

                SJMP  WAIT1

   TX_15H:      MOV   A,#15H         ;向甲机报告呼叫信号不正确

                MOV   SBUF,A

   WAIT2:       JBC   TI,HAVE1

                SJMP  WAIT2

   HAVE1:     
  LJMP  RX_ACK       ;返回到接受呼叫信号状态

   RX_BYTES:    JBC   RI,HAVE2       ;接收数据块长度

                SJMP  RX_BYTES

   HAVE2:       MOV   A,SBUF         ;给长度寄存器赋值

                MOV   R5,A

   RX_NEWS:
  JBC   RI,HAVE3
     ;接收数据

                SJMP  RX_NEWS

   HAVE3:       MOV   A,SBUF
      ;将接受到的数据存入RAM

                MOV   @R1,A

                INC   R1

                ADD   A,R6           ;形成累加和

                MOV   R6,A

                DJNZ  R5,RX_NEWS     ;判断数据是否接收完毕

   RX_SUM:
     JBC   RI,HAVE4         ;接收校验和

                SJMP  RX_SUM

   HAVE4:       MOV   A,SBUF
        ;判断传送是否正确

                SUBB  A,R6

                CJNE  A,#00H,TX_E

   TX_RIGHT:    MOV   A,#0FH         ;向甲机报告传送正确

                MOV   SBUF,A

   WAIT3:       JBC   TI,GOOD



         SJMP  WAIT3

   TX_E:        MOV   A,#0F0H        ;向甲机报告传送有误

                MOV   SBUF,A

   WAIT4:       JBC   TI,AGAIN

                SJMP  WAIT4

   AGAIN:       LJMP  FMT_RAM        ;返回重新接收数据状态

   GOOD:        SJMP  $
实验十一  键盘键值远距离传输实验

一、目的：

利用串行口实现数值的远距离传输。

二、内容：

将A单片机的键值通过串行口传输到B单片机中，并在B单片机的LED显示器上显示；

将B单片机的键值通过串行口传输到A单片机中，并在A单片机的LED显示器上显示。

硬件连接框图见图6-4。

相关电路原理图见第一章图1-9、图1-10。

图 6-4

三、实验步骤：

1、 硬件连接：

1） A单片机实验仪RS-422接口输入端IN（+，-）连接B单片机实验仪RS-

422接口输出端OUT（+，-）；

2） B单片机实验仪RS-422接口输入端IN（+，-）连接A单片机实验仪RS-

422接口输出端OUT（+，-）；

3） 8255（2）片选信号 /CS接 /Y3~/Y7（任选其一）。

4） 连接相关硬件模块的电源短路块。

2、 软件设计：

1） A、B单片机串行口工作方式为方式1。定时器T1做波特率发生器，波

特率为1200HZ（晶振频率为12M）；

2） 8255（2）初始化（扩展显示器及键盘）；

3） 调用显示子程序；

4） 查询串行口接收标志（查询方式），接收到数据即修改显示缓冲区；

5） 调用扫描键盘子程序，

行线输出全零时由列线输入低4位，

列线输出全零时由行线输入高4位，

有键盘输入即将键值从串行口发送；

6） 8255（2）PA口PA7控制小数点，PA6~PA0控制字形（反向驱动），

    8255（2）PB口PB5~PB0控制字位（经反向驱动），

    LED显示器为共阴极；

7） 8255（2）PC口PC7~PC4为行列式键盘的行线，

PC3~PC0 为行列式键盘的列线，

行线、列线均由上拉电阻接+5V；
*四、高级要求：
查询资料，了解其它标准的串行口与8031单片机的连接方式，了解其使用方法，了解相关常用芯片。
五、参考程序：

参考程序中8255（2）片选信号 /CS接 /Y3。

        ORG 0000H

        AJMP START

        ORG 0100H

START:  MOV TMOD,#20H      ;T1定时工作方式2

        MOV TH1,#0CCH       ;T1定时常数(波特率1200）

        MOV TL1,#0CCH

        MOV SCON,#50H      ;串行口方式1,允许接收

        MOV PCON,#80H      ;波特率设置(SMOD=1)

        MOV SP,#30H         ;设置堆栈指针

        SETB TR1            ;启动定时器1

        MOV A,#06H          ;设置显示缓冲区初值

        MOV R0,#75H

STA:    MOV @R0,#10H

        INC R0

        DEC A

        JNZ STA

        LCALL LED         ;调显示子程序

WINA:   JBC RI,CHANGE     ;等待接收

        LCALL KEY         ;调键盘扫描子程序

        LCALL LED

        SJMP WINA

CHANGE: LCALL LED         ;输入串口数据

        CLR RS1

        SETB RS0

        MOV A,SBUF

        MOV R5,A

        MOV R7,#05H        ;将显示缓冲区值右移一位

        MOV R0,#79H

        MOV R1,#7AH

ROT:    MOV A,@R0

        MOV @R1,A

        DEC R0

        DEC R1

        DJNZ R7,ROT

        MOV 75H,R5           ;将接收值存入75H

        AJMP WINA

        ORG 1000H
 KEY:   LCALL LED        ;键盘扫描子程序

        SETB RS1

        CLR RS0

        MOV DPTR,#6003H  ;8255（2）控制字(PC7-PC4输入,PC3-PC0输出)

        MOV A,#88H

        MOVX @DPTR,A

        MOV DPTR,#6002H  ;PC3-PC0输出零,读入PC7-PC4值

        MOV A,#00H

        MOVX @DPTR,A

        MOVX A,@DPTR

        ANL A,#0F0H

        MOV 74H,A           ;将键值的高四位存入74H单元里

        CJNE A,#0F0H,READ   ;无键按下时返回

        RET

READ:   LCALL LED

        SETB RS1

        CLR RS0

        MOV DPTR,#6003H ;8255（2）控制字（PC7-PC4输出,PC3-PC0输入）

        MOV A,#81H

        MOVX @DPTR,A

        MOV DPTR,#6002H   ;PC7-PC4输出零,读入PC3-PC0值

        MOV A,#00H

        MOVX @DPTR,A

        MOVX A,@DPTR

        ANL A,#0FH

        ADD A,74H

        MOV 74H,A          ;将键值存入74H单元里
        ACALL DELAY12MS   ;调用延时程序

 KEY1:  LCALL LED          ;再扫描

        SETB RS1

        CLR RS0

        MOV DPTR,#6003H

        MOV A,#88H

        MOVX @DPTR,A
        MOV DPTR,#6002H

        MOV A,#00H

        MOVX @DPTR,A

        MOVX A,@DPTR

        ANL A,#0F0H

        MOV 73H,A
        LCALL LED

        SETB RS1

        CLR RS0

        MOV DPTR,#6003H

        MOV A,#81H

        MOVX @DPTR,A
        MOV DPTR,#6002H

        MOV A,#00H

        MOVX @DPTR,A

        MOVX A,@DPTR

        ANL A,#0FH

        ADD A,73H

        CLR C              ;如两次值相同，按键未松，再扫描

        SUBB A,74H

        JZ KEY1

        LCALL LED

        SETB RS1

        CLR RS0

        MOV R1,#00H          ;R1作计数器

        MOV R5,#10H          ;十六个键

SEARCH: MOV DPTR,#TABLE1   ;取键值表的首地址

        MOV A,R1

        MOVC A,@A+DPTR     ;取键值表数据

        CLR C

        SUBB A,74H         ;比较读入键值与键值表数据

        JZ SOUT            ;相同转发送程序

        INC R1             ;不相同继续比较

        DJNZ R5,SEARCH

        RET                 ;没有相同的此次无效

SOUT:   MOV A,R1           ;发送键值（0-F）

        MOV SBUF,A

  TR:   LCALL LED

        JBC TI,TR

        RET

LED:    SETB RS1              ;显示子程序

        SETB RS0

        MOV DPTR,#6003H

        MOV A,#80H            ;8255（2）PA口输出PB口输出

        MOVX @DPTR,A
SHOW:   MOV R0,#75h

        MOV R2,#01h

        MOV A,R2
LOOP:   MOV DPTR,#6001H

        MOVX @DPTR,A          ;送位选码到PB口
        MOV DPTR,#TABLE2      ;取显示表的首地址

        MOV A,@R0

        MOVC A,@A+DPTR        ;取字形码
        MOV DPTR,#6000H

        MOVX @DPTR,A           ;送段码到PA口
        ACALL DELAY1MS         ;调用延时程序
        INC R0

        MOV A,R2

        JB ACC.5,LOOP1       ;是否扫描到第6位

        RL A

        MOV R2,A

        AJMP LOOP

LOOP1:  RET

DELAY12MS: MOV R3,#0FFH             ;延时12毫秒

DLAY0:     MOV R4,#0FFH

DLAY1:     DJNZ R4,DLAY1

           DJNZ R3,DLAY0

           RET

DELAY1MS:  MOV R3,#04H               ;延时1毫秒

DELAY0:    MOV R4,#0FFH

DELAY1:    DJNZ R4,DELAY1

           DJNZ R3,DELAY0

           RET

TABLE1: DB 7EH,0BEH,0BDH,0BBH      ;键值表

        DB 0DEH,0DDH,0DBH,0EEH

        DB 0EDH,0EBH,0E7H,0D7H

        DB 0B7H,77H,7BH,7DH,91H

TABLE2: DB 0C0H,0F9H,0A4H,0B0H     ;字型表

        DB 99H,92H,82H,0F8H

        DB 80H,90H,88H,83H

        DB 0C6H,0A1H,86H,8EH,8CH
        END
第七章  D/A实验

实验十二  D/A转换实验（DAC0832）

一、目的

掌握DAC0832与8031的接口方法。掌握D/A转换程序设计方法。

二、内容

将数字量（00H~FFH）从8031传输到DAC0832。测量DAC0832输出电压。

硬件连接框图见图7-1。
    电路原理图见图7-2。

图 7-1

图 7-2
3、 实验步骤：

1、 硬件连接：

DAC0832电路的标准电压端VREF接+5V；

片选信号端 /CS接 /Y3 ~/Y7（任选其一）；

写控制端 /WR接8031的 /WR；

反馈端FG接反馈电阻输入端RFB；

D7 ~ D0接8031的P0.7 ~P0.0；
连接相关硬件模块的电源短路块。
2、 软件设计：

    将数字量（00H~FFH）从8031传输到DAC0832。

传输数据：    MOV  A, #data

                     MOV  DPTR, #addr0832

                   MOVX  @DPTR, A
3、 编程，执行程序；

用万用表测量DAC0832电路的输出端Vout3的输出电压；

在程序中修改输出的数字量，重复测量。
4、 填写以下表格：（自行增加其它数字量）
	数 字 量
	输出电压（V）
	  数 字 量
	 输出电压（V）

	00H
	
	    8FH
	

	0FH
	
	    9FH
	

	1FH
	
	    AFH
	

	2FH
	
	    BFH
	

	3FH
	
	    CFH
	

	4FH
	
	    DFH
	

	5FH
	
	EFH
	

	6FH
	
	FFH
	

	7FH
	
	
	


*四、高级要求：

利用8255（1）扩展DAC0832。

DAC0832的D7~ D0接8255（1）的PA7~PA0，8255（1）和DAC0832的片选信号端 /CS分别接 /Y3 ~/Y7。

数据先传送到8255（1），再从8255（1）传送到DAC0832。
实验十三  程控放大器实验
一、 目的：

了解程控放大器的结构，学会程控放大器的设计和使用。

二、内容：

利用DAC0832设计程控放大器，测量出程控数字量与放大倍数的关系。

硬件连接框图见图7-3。

电路原理图见图7-4。

图7-3

图7-4

三、实验步骤：

1、 硬件连接：

1） +2.5V电源接入电位器WR1一端，另一端接0V（GND）。从中间抽头输出0.1V电压，将此电压接入DAC0832电路的程控放大器输入端（反馈电阻端RFB）。
2） 程控放大器（DAC0832电路）片选信号端 /CS接 /Y3~/Y7（任选其一）；

写控制端 /WR接8031的 /WR；

反馈端FG接标准电压端VREF；

D7~D0接8255（1）PA口的PA7~PA0。

3） 8255（1）片选信号 /CS接 /Y3~/Y7（任选其一）。

4） 连接相关硬件模块的电源短路块。

2、 软件设计：

1） 8255（1）初始化，PA口输出。

2） 将数字量（00H~FFH）送入8255（1）PA口。再将此数字量送入DAC0832。
3、 执行程序：

用万用表测量程控放大器的输出端Vout3的电压。修改程序中输出的数字量，重复以上过程。

4、 填写下列表格，画出坐标图，找出放大器的饱和区。                          

（输入电压0.1V）
	数字量
	输 出 电 压
	放 大 倍 数

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	

	
	
	


*四、高级要求：

1、 将 -5V电源接入电位器WR1一端，另一端接0V（GND）。从中间抽头输

 出 -0.1V电压，将此电压接入DAC0832电路的反馈电阻端RFB（程控放大器

 输入端）。重复以上过程。

2、 分析程控放大器的工作原理。

3、 查询资料，了解其它形式的程控放大器芯片及其使用方法。

五、参考程序框图见下页图7-5：

图 7-5

六、参考程序：
    参考程序中程控放大器（DAC0832电路）片选信号 /CS接 /Y3；

8255（1）片选信号 /CS接 /Y4。

ORG    0000H

                   AJMP   MAIN

                   ORG    0100H

      MAIN:       MOV    DPTR , #6003H         ;8255(1)初始化

                   MOV    A , #80H

                   MOVX   @DPTR , A

        SIN0:      MOV    DPTR , #6000H

                   MOV    A , #00H              ;数据（00H~0FFH）

                   MOVX   @DPTR , A

                   MOV    DPTR , #8000H        ;DAC0832地址

                   MOVX   @DPTR , A

                   AJMP   SIN0

实验十四  程控信号发生器实验

一、目的：

了解程控信号发生器的结构；

学会程控信号发生器的设计和使用。

二、内容：

利用DAC0832的D/A转换功能，根据一定的时序（函数）关系，将相应

的数字量送入DAC0832；

使DAC0832电路的输出端输出相应的波形。

硬件连接框图见图7-6。

电路原理图见图7-2。

图 7-6

三、实验步骤：

1、 硬件连接：

1） DAC0832电路的标准电压端VREF接+5V；

片选信号端 /CS接 /Y3~/Y7（任选其一）；

写控制端 /WR接8031的 /WR；

反馈端FG接反馈电阻输入端RFB；

D7~D0接8255（1）PA口的PA7~PA0。

2） 8255（1）片选信号 /CS接 /Y3~/Y7（任选其一）。

3） 连接相关硬件模块的电源短路块。

2、 软件设计：

1） 8255（1）初始化，PA口输出；

2） 根据一定的时序（函数）关系，将相应的数字量送入8255（1）PA口；

3） 再将此数字量送入DAC0832。

3、 执行程序：

用示波器测量DAC0832电路的输出端Vout3输出的波形。

*四、高级要求：

1、 设计其它波形（如梯形波）的程序。

2、 怎样在程序中处理输出波形的频率？

3、 查询资料，了解专用波形发生器芯片及其使用方法。

五、参考程序框图：

图 7-7(a)  正弦波

图 7-7（b）  三角波

六、参考程序：
参考程序中DAC0832片选信号端 /CS接 /Y3；8255（1）片选信号 /CS接 /Y4。
1） 三角波：

       ORG  0000H

        AJMP MAIN

        ORG  0100H

MAIN:  MOV  DPTR , #8003H     ;8255初始化

        MOV  A , #80H

         MOVX  @DPTR , A
         CLR  A

   UP:  MOV  DPTR , #8000H     ;线性上升段。数据送8255

       MOVX  @DPTR , A
       MOV  DPTR , #6000H     ；数据送DAC0832

       MOVX  @DPTR , A

       ACALL  DING1

       INC  A

       JNZ  UP

       MOV  A , #0FFH
DOWN: DEC A                  ；线性下降段。数据送8255

       MOV  DPTR , #8000H    

       MOVX  @DPTR , A

       MOV  DPTR , #6000H
       MOVX  @DPTR , A       ; 数据送DAC0832

       ACALL  DING1

       JNZ  DOWN

       AJMP  UP

 DING1:   MOV R3,#01H
 DELAY0:  MOV R4,#80H

 DELAY1:  DJNZ R4,DELAY1

           DJNZ R3,DELAY0

           RET  
2） 正弦波：

        ORG  0000H

          AJMP  MAIN

          ORG  0100H

  MAIN:  MOV  DPTR , #8003H    ;8255初始化

          MOV  A , #80H

          MOVX  @DPTR , A

   SIN0:  MOV  R7 , #00H         ;置偏移量

   SIN1:  MOV  A , R7

          MOV  DPTR , #TABLE    ;从表中取值

          MOVC  A , @A+DPTR

          MOV   DPTR , #8000H    ; 数据送8255

          MOVX  @DPTR , A       
          MOV   DPTR , #6000H    ；数据送DAC0832

          MOVX  @DPTR , A

          INC    R7               ；指向下一个数

          CJNE  R7 , #0B5H , SIN1   ；一周期180个数（B5H=181D）

          AJMP  SIN0

TABLE:  DB 7FH,83H,87H,8CH,90H,95H,99H,9DH,0A2H；（每2度一个数）
         DB 0A6H,0AAH,0AEH,0B2H,0B6H,0BAH,0BEH,0C2H

         DB 0C6H,0C9H,0CDH,0D0H,0D3H,0D7H,0DAH,0DDH

         DB 0E0H,0E3H,0E5H,0E8H,0EAH,0EDH,0EFH,0F1H

         DB 0F3H,0F4H,0F6H,0F7H,0F9H,0FAH,0FBH,0FCH

         DB 0FCH,0FDH,0FDH,0FDH,0FEH,0FDH,0FDH,0FDH

         DB 0FCH,0FCH,0FBH,0FAH,0F9H,0F7H,0F6H,0F4H

         DB 0F3H,0F1H,0EFH,0EDH,0EAH,0E8H,0E5H,0E3H

         DB 0E0H,0DDH,0DAH,0D7H,0D4H,0D0H,0CDH,0C9H
         DB 0C6H,0C2H,0BEH,0BAH,0B6H,0B2H,0AEH,0AAH

         DB 0A6H,0A2H,9DH,99H,95H,90H,8CH,88H,83H

         DB 7FH,7AH,76H,71H,6DH,69H,64H,60H,5CH,57H

         DB 53H,4FH,4BH,47H,43H,3FH,3BH,38H,34H,31H

         DB 2DH,2AH,26H,23H,20H,1DH,1BH,18H,15H,13H

         DB 11H,0EH,0CH,0BH,09H,07H,06H,04H,03H,02H,01H

         DB 01H,00H,00H,00H,00H,00H,00H,00H,01H,01H

         DB 02H,03H,04H,06H,07H,09H,0AH,0CH,0EH,10H

         DB 13H,15H,18H,1AH,1DH,20H,23H,26H,29H,2DH

         DB 30H,34H,37H,3BH,3FH,43H,46H,4BH,4FH,53H

         DB 57H,5BH,5FH,64H,68H,6CH,71H,75H,7AH,7FH

第八章  A/D转换实验

实验十五  A/D转换实验（ADC0809）

一、目的：

掌握ADC0809和MCS-51单片机接口设计方法；

了解8031读取ADC0809转换结果的程序设计方法。

二、内容：

将直流电压从ADC0809的输入端输入，转换成对应的数字量后，送LED

 显示器显示。显示范围：00H~FFH。

硬件连接框图见图8-1。

相关电路原理图见第一章图1-6和图1-9。

图 8-1

三、实验步骤：

1、 硬件连接：

1） 将+5V电压接入电位器WR1的一端，0V（GND）接入另一端。中间抽头输出0V~5V，此电压接入ADC0809输入端IN0；
2） ADC0809读、写信号控制端 /RD、/WR接8031的 /RD、/WR；

ADC0809时钟信号CLOCK接1/2ALE；

ADC0809片选信号 /CS接 /Y3~/Y7（任选其一）；

ADC0809输出D7~D0接8031的P0.7~P0.0；

3） 8255(2) 片选信号 /CS接 /Y3~/Y7（任选其一）；

4） 连接相关硬件模块的电源短路块。

2、 软件设计：

1） 8255（2）初始化（PA口、PB口输出，扩展LED显示器）。

2） 8255（2）PA口PA7控制小数点，PA6~PA0控制字形（反向驱动），PB口PB5~PB0控制字位（反向驱动）。LED显示器为共阴极。

3） 数据采集采用延时方式，驱动A/D转换后，延时150μS左右，读取数据。

*四、高级要求：

1、 将采集的数据变换成BCD码后送显示器显示。

2、 对两路信号进行采集：在以上实验的基础上，将+5V电压接入电位器WR2的一端，0V（GND）接入另一端，中间抽头输出0V~5V，此电压接入ADC0809输入端IN1。IN0采集的数据在LED显示器左边显示，IN1采集的数据在LED显示器右边显示。

3、 数据采集采用查询及中断方式，利用 /EOC信号进行数据采集。

4、 参考程序：

参考程序中8255(2)片选信号接/Y3，ADC0809片选信号接/Y4。

           ORG 0000H

          AJMP START

          ORG 0100H

   START: MOV R0,#50H

          MOV R2,#06H

   LOOP0:MOV @R0,#10H             ;给显示缓冲区50H-55H赋值

          INC R0                     ;显示全黑

          DJNZ R2,LOOP0

          MOV DPTR,#6003H

          MOV A,#80H

          MOVX @DPTR,A     ;对8255(2)进行初始化(接LED显示器)

    MAIN:ACALL LED                  ;调显示子程序。

          MOV DPTR,#8000H            ;ADC0809地址

          MOV A,#00H                 ;启动通道0(IN0)

          MOVX @DPTR,A

          MOV R7,#20H               ;延时150us(CLK接1/2ALE)

     L1:  NOP

          NOP

          NOP

          DJNZ R7,L1

          MOVX A,@DPTR              ;读数据

          MOV B,A

          ANL A,#0F0H

          SWAP A

          MOV 50H,A

          MOV A,B

          ANL A,#0FH

          MOV 51H,A

          AJMP MAIN

     LED: MOV R0,#01H                ;位码

          MOV R1,#50H                ;显示缓冲区首址

    LOOP: MOV A,R0                   ;送位码

          MOV DPTR,#6001H

          MOVX @DPTR,A

          MOV A,@R1                  ;送段码

          MOV DPTR,#SEG

          MOVC A,@A+DPTR

          MOV DPTR,#6000H

          MOVX @DPTR,A

          ACALL DELAY

          MOV A,#7FH                 ;清黑

          MOVX @DPTR,A

          INC R1

          MOV A,R0                   ;指向下一位

          RL A

          MOV R0,A

          CJNE R0,#04H,LOOP          ;两位显示完返回

          RET

   DELAY: MOV R7,#02H

  DELAY0: MOV R6,#0FFH

  DELAY1: DJNZ R6,DELAY1

          DJNZ R7,DELAY0

          RET

     SEG: DB 40H,79H,24H,30H,19H,12H,02H,78H  ;字形表

          DB 00H,10H,08H,03H,46H,21H,06H,0EH

实验十六  A/D转换实验（ICL7135）

一、目的：

掌握四位半A/D芯片ICL7135和MCS-51单片机接口设计方法。

了解8031读取ICL7135转换结果的程序设计方法。

二、内容：

将直流电压从ICL7135的输入端输入，转换成对应的数字量后，送LED

显示器显示。显示范围：00000-19999。

硬件连接框图见图8-2

相关电路原理图见第一章图1-7、图1-8和图1-9。

图 8-2

三、实验步骤：

1、 硬件连接：

1） 将+2.5V电压接入电位器WR1的一端，0V（GND）接入另一端。中间抽

头输出0V~2.5V，此电压接入ICL7135输入端Vin5；

2） ICL7135数据输出端B8~B1（BCD码输出）及位状态输出端D5~D1（万、千、百、十、个位输出状态）分别接8255（1）PA口、PB口；

3） ICL7135的时钟输入端CLKIN接1/8 ALE，转换状态输出端BUSY接8031

的外部中断输入端 /INT0；

4） 8255（1）片选信号 /CS接 /Y3~/Y7（任选其一）；

    8255（2）片选信号 /CS接 /Y3~/Y7（任选其一）；

5） 连接相关硬件模块的电源短路块。

2、 软件设计：

1） 8255（1）初始化（PA口、PB口输入，连接ICL7135）；

8255（2）初始化（PA口、PB口输出，扩展LED显示器）；

设置外部中断 /INT0（用BUSY申请中断），低电平有效。

2） 主程序调用显示程序，并循环等待中断。

3） 中断服务子程序判别万、千、百、十、个位输出状态，分别读入万、千、百、十、个位数据并保存。修改显示缓冲器内容。

4） 8255（2）PA口PA7控制小数点，PA6~PA0控制字形（反向驱动），PB口PB5~PB0控制字位（反向驱动）。LED显示器为共阴极。
*四、高级要求：

1、 怎样处理ICL7135的溢出标志位OVER及欠量程标志位UNDER？修改硬

件连接及程序设计，过量程、欠量程时在LED显示器上提示。

2、 查询资料，了解ICL7135与8031单片机二线式连接法（利用A/D转换的

时间长短计算出A/D数值）。

五、参考程序：
参考程序中：

ICL7135数据输出端B8~B1（BCD码输出）接8255（1）PA口的PA7~PA4，ICL7135位状态输出端D4~D1（千、百、十、个位输出状态）接8255（1）PA口的PA3~PA0；

    查询万位时，无千、百、十、个位输出，即为万位输出；

    ICL7135的时钟输入端CLKIN接1/8 ALE，转换状态输出端BUSY接8031

的外部中断输入端 /INT0；

    8255（1）片选信号 /CS接 /Y7；

    8255（2）片选信号 /CS接 /Y3。

          ORG 0000H

          AJMP START

          ORG 0003H            ;INT0中断矢量地址入口

          LJMP INTER

          ORG 0100H

   START: MOV R0,#50H

          MOV R2,#06H

   LOOP0: MOV @R0,#10H         ;给显示缓冲区50H-55H赋值

          INC R0                 ;显示全黑

          DJNZ R2,LOOP0

          MOV DPTR,#6003H      ;对8255(2)进行初始化(LED显示器)

          MOV A,#80H

          MOVX @DPTR,A        

          MOV A,#92H            ;对8255(1)进行初始化(接ICL7135)

          MOV DPTR,#0E003H

          MOVX @DPTR,A

          MOV SP,#70H            ;设置堆栈区

          SETB IT0                ;/INT0 跳变触发

          SETB EA

          SETB EX0                ;开中断

    MAIN: ACALL LED              ;调显示子程序

          NOP                     ;等待中断

          NOP

          AJMP MAIN

     LED: CLR EA                     ;关中断

          MOV R0,#01H                ;位码

          MOV R1,#50H                ;显示缓冲区首址

    LOOP: MOV A,R0                   ;送位码

          MOV DPTR,#6001H

          MOVX @DPTR,A

          MOV A,@R1                  ;送段码

          MOV DPTR,#SEG

          MOVC A,@A+DPTR
          MOV DPTR,#6000H

          MOVX @DPTR,A

          ACALL DELAY

          MOV A,#7FH                 ;清黑

          MOVX @DPTR,A

    GOON: INC R1                     ;指向下一位

          MOV A,R0 

          RL A

          MOV R0,A

          CJNE R0,#40H,LOOP          ;六位显示完返回

          SETB EA                    ;开中断

          RET

   DELAY: MOV R7,#02H

  DELAY0: MOV R6,#0FFH

  DELAY1: DJNZ R6,DELAY1

          DJNZ R7,DELAY0

          RET

          ORG 1000H

   INTER: PUSH PSW                   ;中断子程序

          PUSH DPH                   ;现场保护

          PUSH DPL

          PUSH ACC
          CLR EA                     ;关中断

     D5:  MOV DPTR,#0E000H

          MOVX A,@DPTR               ;等待万位输出                                       

          MOV R3,A

          ANL A,#0FH

          JNZ D5

          MOV A,R3

          ANL A,#0F0H              ;万位BCD码放入50H

          SWAP A

          MOV 50H,A

     D4:  MOV DPTR,#0E000H

          MOVX A,@DPTR

          MOV R3,A

          ANL A,#08H            ;等待千位输出
          JZ D4

          MOV A,R3

          ANL A,#0F0H

          SWAP A

          MOV 51H,A             ;千位BCD码放入51H

     D3:  MOV DPTR,#0E000H      ;等待百位输出

          MOVX A,@DPTR

          MOV R3,A

          ANL A,#04H

          JZ D3

          MOV A,R3

          ANL A,#0F0H

          SWAP A

          MOV 52H,A              ;百位BCD码放入52H

     D2:  MOV DPTR,#0E000H       ;等待十位输出

          MOVX A,@DPTR

          MOV R3,A

          ANL A,#02H

          JZ D2

          MOV A,R3

          ANL A,#0F0H

          SWAP A

          MOV 53H,A                ;十位BCD码放入53H

     D1:  MOV DPTR,#0E000H         ;等待个位输出

          MOVX A,@DPTR

          MOV R3,A

          ANL A,#01H

          JZ D1

          MOV A,R3

          ANL A,#0F0H

          SWAP A

          MOV 54H,A                ;个位BCD码放入54H

  RETURN: POP ACC                   ;恢复现场

          POP DPL

          POP DPH

          POP PSW

          SETB EA                   ;开中断

          RETI

     SEG: DB 40H,79H,24H,30H,19H,12H,02H,78H

          DB 00H,10H,08H,03H,46H,21H,06H,0EH

          DB 7FH                            ；7FH为全黑字形码

第九章  数据处理程序设计实验
实验十七  数字滤波程序设计实验

一、实验目的：

    了解数据采集中数字滤波的方法，掌握数字滤波的程序设计。

二、实验内容：

采集的数据已存放在指定的存储单元，采用去极值平均滤波法（去掉最

 小值及最大值，然后求平均值），求得有效值。

三、实验步骤：

1、 软件设计说明：

    假设己采集到六个数据（六个数据在程序中赋值给出），每个数据为两字节十六进制数（0000H~FFFFH）。

    数据存放的地址为：(2BH,2AH)、(29H,28H)、(27H,26H)、(25H,24H)、(23H,22H)、(21H,20H)。高字节在前。

将六个数排序后，去掉最小值及最大值，然后求平均值。将此平均值作为 

有效值存放在（61H,60H）。

2、执行程序，暂停，在窗口中检查结果。修改程序中给定的六个数据，重复以上过程。

四、数字滤波简介：

1、 程序判断滤波

在生产实践中，很多物理量的变化有一定的规律和需要一定的时间，相邻两次采样值之间的变化也有一个限额。为此，可根据经验数据，规定出一个最大可能的变化范围，每次采样值均与上次的有效采样值进行比较。如果变化幅值不超过规定值限，则本次采样值有效，否则，本次采样值应视为干扰而弃之，以上次采样值为准。

2、 中值滤波

    对目标参数连续进行若干次采样，然后将这些采样进行排序，选取中间位置的采样值为有效值。本算法为取中值，采样次数应为奇数。常用3次或5次。对于变化很缓慢的参数，有时也可增加次数。

3、 算术平均滤波

对于目标参数进行连续采样，然后求其算术平均值作为有效采样值。该算法适用于抑制随机干扰。采样次数n越大，平滑效果越好，但系统的灵敏度要下降。为方便求平均值，n一般取4、8、16类的2的整数幂，以便用位移来代替除法。

4、 去极值平均滤波

    连续采样n次，去除其中的最大值和最小值，按n-2个采样值求平均，即

得有效值。为使平均滤波方便，n-2应为2、4、8、16，故n常取4、6、10、

18。具体做法有两种：对于快变参数，先连续采样n次，然后再处理，但要在

RAM中开辟出n个数据的暂存区；对于慢变参数，可一边采样，一边处理，而不必在RAM中开辟数据暂存区。

*五、高级要求：
假设己采集到六个数据（六个数据在程序中赋值给出），每个数据为一字节十六进制数（00H~FFH）。

数据存放的地址为：25H、24H、23H、22H、21H、20H。

将六个数排序后，去掉最小值及最大值，然后求平均值。将此平均值作为有效值存放在60H。
六、参考程序：
    平均值在(23H,22H)中。

          ORG 0000H

          AJMP MAIN

          ORG 0100H

MAIN: MOV 2BH，#XXH    ;给(2BH,2AH) ~ (21H,20H)赋值

      MOV 2AH，#XXH

      MOV 29H，#XXH

      MOV 28H，#XXH


   MOV 27H，#XXH

      MOV 26H，#XXH

      MOV 25H，#XXH

      MOV 24H，#XXH

      MOV 23H，#XXH

      MOV 22H，#XXH

      MOV 21H，#XXH

      MOV 20H，#XXH

          LCALL MAX

          LCALL MIN

          LCALL JIA

          SJMP $

    MAX:  CLR C         ;(2BH,2AH) ~ (21H,20H)最大值放入(21H,20H)

          MOV A,20H

          SUBB A,22H

          MOV A,21H

          SUBB A,23H

          JC APUT1

          LJMP AXX1

   APUT1: MOV A,20H

          MOV 20H,22H

          MOV 22H,A

          MOV B,21H

          MOV 21H,23H

          MOV 23H,B

    AXX1: CLR C

          MOV A,20H

          SUBB A,24H

          MOV A,21H

          SUBB A,25H

          JC APUT2

          LJMP AXX2

   APUT2: MOV A,20H

          MOV 20H,24H

          MOV 24H,A

          MOV B,21H

          MOV 21H,25H

          MOV 25H,B

    AXX2: CLR C

          MOV A,20H

          SUBB A,26H

          MOV A,21H

          SUBB A,27H

          JC APUT3

          LJMP AXX3

   APUT3: MOV A,20H

          MOV 20H,26H

          MOV 26H,A

          MOV B,21H

          MOV 31H,27H

          MOV 27H,B

    AXX3: CLR C

          MOV A,20H

          SUBB A,28H

          MOV A,21H

          SUBB A,29H

          JC APUT4

          LJMP AXX4

   APUT4: MOV A,20H

          MOV 20H,28H

          MOV 28H,A

          MOV B,21H

          MOV 21H,29H

          MOV 29H,B

    AXX4: CLR C

          MOV A,20H

          SUBB A,2AH

          MOV A,21H

          SUBB A,2BH

          JC APUT5

          LJMP AXX5

   APUT5: MOV A,20H

          MOV 20H,2AH

          MOV 2AH,A

          MOV B,21H

          MOV 21H,2BH

          MOV 2BH,B

    AXX5: RET

     MIN: CLR C        ;(2BH,2AH) ~ (23H,22H)最小值放入(2BH,2AH)

          MOV A,2AH

          SUBB A,22H

          MOV A,2BH

          SUBB A,23H

          JNC PUT11

          AJMP XX1

   PUT11: MOV A,2AH

          MOV 2AH,22H

          MOV 2AH,A

          MOV B,2BH

          MOV 2BH,23H

          MOV 23H,B

     XX1: CLR C

          MOV A,2AH

          SUBB A,24H

          MOV A,2BH

          SUBB A,25H

          JNC PUT22

          AJMP XX2

   PUT22: MOV A,2AH

          MOV 2AH,24H

          MOV 24H,A

          MOV B,2BH

          MOV 2BH,25H

          MOV 25H,B

     XX2: CLR C

          MOV A,2AH

          SUBB A,26H

          MOV A,2BH

          SUBB A,27H

          JNC PUT33

          AJMP XX3

   PUT33: MOV A,2AH

          MOV 2AH,26H

          MOV 26H,A

          MOV B,2BH

          MOV 2BH,27H

          MOV 27H,B

     XX3: CLR C

          MOV A,2AH

          SUBB A,28H

          MOV A,29H

          JNC PUT44

          AJMP XX4

   PUT44: MOV A,2AH

          MOV 2AH,28H

          MOV 28H,A

          MOV B,2BH

          MOV 2BH,29H

          MOV 29H,B

     XX4: RET

     JIA: MOV A,22H   ;(29H,28H) ~ (23H,22H)平均值放入(23H,22H)

          ADD A,24H

          MOV 22H,A

          MOV A,23H

          ADDC A,25H

          MOV 23H,A

          MOV A,22H

          ADD A,26H

          MOV 22H,A

          MOV A,23H

          ADDC A,27H

          MOV 23H,A

          MOV A,22H

          ADD A,28H

          MOV 22H,A

          MOV A,23H

          ADDC A,29H

          MOV 23H,A

          MOV A,23H

          RRC A

          MOV 23H,A

          MOV A,22H

          RRC A

          MOV 22H,A

          CLR C

          MOV A,23H

          RRC A

          MOV 23H,A

          MOV A,22H

          RRC A

          MOV 22H,A

          RET

实验十八  非线性校正及标度变换程序设计实验

一、实验目的：

    了解非线性校正及标度变换的方法，掌握非线性校正及标度变换的程序设计。

二、实验内容：

    根据表格中给出的数据，采用等节点分段直线插值法，建立非线性校正及标度变换公式（温度与A/D数值之间的关系式）。

    根据给出的A/D采集数据（存放在61H、60H单元的两字节16进制数，由程序中赋值给出，范围为：0000H ~ FFFFH ），利用非线性校正及标度变换公式求出温度值（温度范围为00.0℃ ~ 99.9℃）。

将温度值在LED显示器上显示。

三、实验步骤：
1、 硬件连接:

8255(2)片选信号 /CS连接 /Y3 ~ /Y7（任选其一）。

连接相关硬件模块的电源短路块。

LED显示器电路原理图见第一章图1-9。

2、 软件设计说明：

1） 显示子程序的设计：

显示温度值为三位十进制数，范围为00.0℃ ~ 99.9℃(小数点在确定个位数的字型码时给出)。在右边三位LED显示器上显示。

    8255（2）PA口的PA7控制小数点，PA6 ~ PA0控制字形段选（反向驱动），PB口PB5~PB0控制字位（反向驱动）。

    共阴极LED显示器。

2） 建立表格：

    已知11个测量点的采集数据见下表：
	温度（℃）
	  0.0
	10.0
	20.0
	30.0
	40.0
	50.0

	A/D数值
	  00195
	02300
	04319
	06355
	8302
	10285

	十六进制
	00C3H
	08FCH
	10DFH
	18D3H
	206EH
	282DH

	
	
	
	
	
	
	

	温度（℃）
	60.0
	70.0
	80.0
	90.0
	99．9
	

	A/D数值
	12256
	14201
	16161
	18025
	19627
	

	十六进制
	2FE0H
	3779H
	3F21H
	4669H
	4CABH
	


3） 建立插值公式：

将上表按温度均匀分成十段，以温度值和A/D采集数据为函数关系建立十段线性插值公式。

    设直线方程y = ax+b。y是温度值，x是A/D采集数据。

    从表中可以得出每段的（x1,y1）和(x2,y2)。

    求解方程 y1 = ax1+b           则 a = (y2-y1) / (x2-x1)

             y2 = ax2+b              b = (x1y2-x2y1) / (x1-x2)

4） 建立插值公式举例：

从表中可知，0℃时A/D采集数据为00195D（00C3H）,10℃时A/D采集数据为02300D (08FCH)。

则  y1=000        x1=195

        y2=100        x2=2300           （y为温度值，最后一位为小数）
根据以上方法求得1/a ≈ 21.1D, b ≈ -1D。
    为了提高测量精度，设m = 1/a×10＝211D＝D3H, n = - b = 1D = 01H。

    则0℃ ~ 10℃的插值公式y=ax+b修改为：y=(10×x)/m - n。           

5） 非线性校正及标度变换：
    根据A/D采集数据x的大小（由程序中赋值给出，存放在61H、60H单元）。选择插值公式，求出被测温度值y。

6） 16进制数→十进制数转换程序设计：

    根据插值公式求得的被测温度y为两字节16进制数，将其转换成十进制数（范围为000D—999D），再将其送入显示缓冲区。小数点在确定个位数的字型码时给出。显示范围为00.0℃ ~ 99.9℃。
7） 当x数值超出00C3H ~ 4CABH范围时，给出欠量程或过量程标志（欠量程显示全“－”，过量程显示全“H” ）。

3、执行程序，观察结果。修改程序中的x值，重复以上过程。
*四、高级要求：
根据给出的A/D采集数据（存放在60H单元的一字节16进制数，由程序中赋值给出，范围为：00H ~ FFH ），利用非线性校正及标度变换公式求出温度值（温度范围为00.0℃ ~ 99.9℃）。

将温度值在LED显示器上显示。

已知11个测量点的采集数据见下表：
	温度（℃）
	  0.0
	10.0
	20.0
	30.0
	40.0
	50.0

	A/D数值
	  
	
	
	
	
	

	十六进制
	
	
	
	
	
	

	
	
	
	
	
	
	

	温度（℃）
	60.0
	70.0
	80.0
	90.0
	99．9
	

	A/D数值
	
	
	
	
	
	

	十六进制
	
	
	
	
	
	


五、参考程序:

参考程序中8255(2)片选信号 /CS连接 /Y3。

          ORG 0000H

          AJMP MAIN

          ORG 0100H

    MAIN: MOV A,#XXH        ;给(61H,60H)赋值 (A/D采集数值)

          MOV 61H,A

          MOV A,#XXH

          MOV 60H,A

          MOV R0,#56H

          MOV R2,#03H

    PUT1: MOV @R0,#10H      ;给显示缓冲区(58H-56H)赋值显示全黑

          INC R0

          DJNZ R2,PUT1

          MOV DPTR,#6003H   ;对8255(2)初始化(接LED显示器)

          MOV A,#81H

          MOVX @DPTR,A      
    PUT2: ACALL CHANGE      ;调非线性校及标度变换程序

          ACALL LED         ;调显示子程序

          NOP

          NOP

          AJMP PUT2

     LED: MOV R0,#20H             ;显示子程序

          MOV R1,#56H

    LOOP: MOV A,R0                ;送位码(PB口)

          MOV DPTR,#6001H

          MOVX @DPTR,A

          MOV A,@R1               ;送段码(PA口)

          MOV DPTR,#SEG

          MOVC A,@A+DPTR

          CJNE R0,#10H,LOOP1

     DOT: ADD A,#80H

          MOV DPTR,#6000H        ;个位带小数点

          MOVX @DPTR,A

          AJMP GOON

   LOOP1: MOV DPTR,#6000H        ;非个位不带小数点

          MOVX @DPTR,A

   GOON:  ACALL DELAY

          MOV A,#7FH

          MOVX @DPTR,A            ;清黑

          MOV A,R0

          RR A

          MOV R0,A

          INC R1

          CJNE R0,#04H,LOOP       ;三位显示完?

          RET
   DELAY: MOV R7,#02H

  DELAY0: MOV R6,#0FFH

  DELAY1: DJNZ R6,DELAY1

          DJNZ R7,DELAY0

          RET

  CHANGE: MOV R0,60H   ;非线性校及标度变换(含16进制→BCD码)

          MOV R1,61H   ;按温度分段

          CJNE R1,#00H,L1   ;与0℃(00C3H)比较。不等转至L1

          CJNE R0,#0C3H,L2  ;                  不等转至L2

CHA1: MOV A,#00H        ;0℃~10℃          

AJMP GO

      L1: CJNE R1,#08H,L3   ;与10℃(08FCH)比较。不等转至L3

          CJNE R0,#0FCH,L4  ;                   不等转至L4

    CHA2: MOV A,#02H        ;10℃~20℃

          AJMP GO

      L2: JC ERROR1         ;即小于0℃时转 ERROR1

          AJMP CHA1         ;即高位为#00H,必在0℃~10℃之间

  ERROR1: MOV 56H,#11H      ;三位全部显示“-”

          MOV 57H,#11H

          MOV 58H,#11H

          AJMP QU2

L3: JC CHA1           ;C=1,(61H)<#08H,必在0℃~10℃

          CJNE R1,#10H,L5   ;与20℃(10DFH)比较。

          CJNE R0,#0DFH,L6

    CHA3: MOV A,#04H        ;20℃~30℃

          AJMP GO

      L4: JC CHA1           ;C=1,(61H)=08H,(60H)<FCH,必在0℃~10℃

          AJMP CHA2         ;C=0,(61H)=08H,(60H)>FCH,必在10℃~20℃

L5: JC CHA2           ;C=1,(61H)<10H,必在10℃~20℃

          CJNE R1,#18H,L7   ;与30℃(18D3H)比较。

          CJNE R0,#0D3H,L8

    CHA4: MOV A,#06H        ;30℃~40℃

          AJMP GO

      L6: JC CHA2         ;C=1,(61H)=10H,(60H)<DFH,必在10℃~20℃

          AJMP CHA3       ;C=0,(61H)=10H,(60H)>DFH,必在20℃~30℃

L7: JC CHA3

          CJNE R1,#20H,L9       ;与40℃(206EH)比较。

          CJNE R0,#6EH,L10

    CHA5: MOV A,#08H            ;40℃~50℃

          AJMP GO

      L8: JC CHA3

          AJMP CHA4

L9: JC CHA4

          CJNE R1,#28H,L11      ;与50℃(282DH)比较。

          CJNE R0,#2DH,L12

    CHA6: MOV A,#0AH            ;50℃~60℃

          AJMP GO

     L10: JC CHA4

          AJMP CHA5

L11: JC CHA5

          CJNE R1,#2FH,L13      ;与60℃(2FE0H)比较。

          CJNE R0,#0E0H,L14

    CHA7: MOV A,#0CH            ;60℃~70℃

          AJMP GO

     L12: JC CHA5

          AJMP CHA6

L13: JC CHA6

          CJNE R1,#37H,L15      ;与70℃(3779H)比较。

          CJNE R0,#79H,L16

    CHA8: MOV A,#0EH            ;70℃~80℃

          AJMP GO

     L14: JC CHA6

          AJMP CHA7

L15: JC CHA7

          CJNE R1,#3FH,L17      ;与80℃(3F21H)比较。

          CJNE R0,#21H,L18

    CHA9: MOV A,#10H            ;80℃~90℃

          AJMP GO

     L16: JC CHA7

          AJMP CHA8

L17: JC CHA8

          CJNE R1,#46H,L19      ;与90℃(4669H)比较。

          CJNE R0,#69H,L20

   CHA10: MOV A,#12H            ;90℃~99.9℃

          AJMP GO

     L18: JC CHA8

          AJMP CHA9

L19: JC CHA9

          CJNE R1,#4CH,L21     ;与99.9℃(4CABH)比较。

          CJNE R0,#0ABH,L22

          AJMP CHA10          ;恰好为99.9℃时转CHA10

     L20: JC CHA9

          AJMP CHA10

     L21: JC CHA10

          AJMP ERROR2

L22: JC CHA10

  ERROR2: MOV 56H,#12H          ;三位全部显示“H”

          MOV 57H,#12H

          MOV 58H,#12H

          AJMP QU2

      GO: MOV R5,A        ;计算y=(10×x)/m - n

          MOV DPTR,#TABLE

          MOVC A,@A+DPTR

          MOV 37H,A       ;查表将m放入37H(除数)

          MOV A,R5

          INC A

          MOVC A,@A+DPTR

          MOV 38H,A       ;查表将n放入38H

  MULKLM: CLR C

          MOV A,#0AH      ;计算 y=10×x

          MOV B,60H       ;X为(61H,60H)

          MUL AB

          MOV 67H,A       ;积为(69H,68,67H)

          MOV 68H,B

          MOV A,#0AH

          MOV B,61H

          MUL AB

          ADDC A,68H

          MOV 68H,A

          MOV A,B

          ADDC A,#00H

          MOV 69H,A

          MOV 43H,#00H    ;计算(10×x)/m

          MOV 42H,#00H    ;被除数(69H,68H,67H) 除数(37H)

          MOV 62H,#00H    ;商(43H,42H) 余数(62H)

    DIVD: CLR C

          MOV A,67H

          SUBB A,37H

          MOV B,A

          JC QUIT1       ;不够减的话转至 QUIT1

          MOV 67H,A

          AJMP G         ;够减的话转至 G

   QUIT1: MOV A,68H

          JZ QUIT3  ;最低位不够减且第二位为0,向最高位借位转QUIT3

   QUIT2: MOV 67H,B ;最低位不够减但第二位不为0时

          MOV A,68H

          SUBB A,#00H

          MOV 68H,A

          AJMP G     ;此部分处理完毕后转至 G

   QUIT3: MOV A,69H

          JZ SUBB1   ;借位但最高位和第二位都为0时除法转换结束

          SUBB A,#00H

          MOV 69H,A

          MOV 68H,#0FFH

          MOV A,#0FFH

          SUBB A,37H

          ADD A,67H

          INC A

          MOV 67H,A

       G: MOV 62H,67H        ;62H放余数

          MOV A,42H

          ADD A,#01H

          MOV 42H,A

          MOV A,43H

          ADDC A,#00H

          MOV 43H,A

          AJMP DIVD

   SUBB1: MOV A,62H          ;62H里放余数

          CJNE A,#5DH,SUBB2  ;不等转SUBB2,进行下一步判别

          JC SUBB3           ;相等(5入),跳转至SUBB3

   SUBB2: JC SUBB4           ;(62H)<5DH,(4舍)并跳转至SUBB4

   SUBB3: CLR C

          MOV A,42H

          ADD A,#01H

          MOV 42H,A

          MOV A,43H

          ADDC A,#00H

          MOV 43H,A

   SUBB4: CLR C              ;计算 (10×x/m) - n

          MOV A,42H

          SUBB A,38H

          MOV 42H,A

          MOV A,43H

          SUBB A,#00H

          MOV 43H,A

    HTOB: MOV 56H,#00H        ;16进制(43H,42H)→BCD码

          MOV 57H,#00H

          MOV 58H,#00H        ;BCD码:5AH(千位) ~ 56H(小数位)

          MOV 59H,#00H      

          MOV 5AH,#00H

   HWWEI: CLR C               ;处理千位

          MOV A,42H

          SUBB A,#10H

          MOV B,A

          MOV A,43H           ;2710H=10000B

          SUBB A,#27H

          JC HQWEI

          INC 5AH

          MOV 43H,A

          MOV 42H,B

          AJMP HWWEI

   HQWEI: CLR C                ;处理百位

          MOV A,42H

          SUBB A,#0E8H

          MOV B,A

          MOV A,43H            ;03E8H=1000B

          SUBB A,#03H

          JC HBWEI

          INC 59H

          MOV 43H,A

          MOV 42H,B

          AJMP HQWEI

   HBWEI: CLR C                 ;处理十位

          MOV A,42H

          SUBB A,#64H

          MOV B,A

          MOV A,43H             ;64H=100B

          SUBB A,#00H

          JC HSWEI

          INC 58H

          MOV 43H,A

          MOV 42H,B

          AJMP HBWEI

   HSWEI: CLR C                  ;处理个位

          MOV A,42H

          SUBB A,#0AH

          JC HGWEI               ;0AH=10B

          INC 57H

          MOV 42H,A

          AJMP HSWEI

   HGWEI: MOV 56H,42H            ;处理小数位

     QU2: RET

   TABLE: DB 0D3H,09H,0CAH,09H,0CCH,0CH,0C3H,1AH

          DB 0C6H,13H,0C5H,16H,0C3H,1EH,0C4H,19H

          DB 0BAH,43H,0A2H,0D6H                      ;参数m,n

     SEG: DB 40H,79H,24H,30H,19H,12H,02H,78H,00H

          DB 10H,08H,03H,46H,21H,06H,0EH,7FH,3FH,09H ;段码
第十章  测量与控制实验

实验十九  频率测量实验

一、目的：

了解利用单片机测量频率的方法。

二、内容：

利用单片机测量555振荡电路输出的信号频率。

测量数据在LED显示器上显示。

硬件连接框图见图10-1。

相关电路见第一章图1-9、图1-12。

三、实验步骤：

1、 硬件连接：

555振荡电路输出Fout接8031定时器/计数器0输入端T0；

8255（2）片选信号/CS接/Y3~/Y7（任选其一）；

连接相关硬件模块的电源短路块。

2、 软件设计：

定时器/计数器T0计数；

定时器/计数器T1进行50MS定时中断；

中断响应后读入T0数据；

数据乘以20后转换成BCD码送去显示（以KHZ为单位）。

3、 转动WR8，观察555振荡电路输出信号的频率范围。

*四、高级要求：

1、 8051单片机（12M晶振）测量频率的范围？

2、 利用单片机测量555振荡电路输出的信号周期。

测量数据在LED显示器上显示。

五、参考程序：

参考程序中8255（2）片选信号/CS接/Y3。

（单位为KHZ）

        ORG 0000H

        AJMP MAIN              

        ORG 001BH              ;T1中断子程序入口

        AJMP T11

        ORG 0100H

MAIN:  MOV DPTR,#6003H       ;对8255(2)进行初始化(接LED显示器)

        MOV A,#80H

        MOVX @DPTR,A

        MOV TMOD,#15H         ;T0计数、T1定时50MS

        MOV TH0,#00H

        MOV TL0,#00H

        MOV TH1,#3CH

        MOV TL1,#0B0H

        SETB EA                ;开CPU、T1中断，关T0中断

        SETB ET1

        CLR ET0

        MOV R7,#06H            ;显示缓冲区75H-70H清零

        MOV R0,#70H

L0:     MOV @R0,#00H

        INC R0

        DJNZ R7,L0

        SETB TR0                ;启动T0、T1

        SETB TR1

L1:     ACALL LED              ;等待T1中断

        NOP

        NOP

        NOP

        SJMP L1

        ORG 0200H

T11:    CLR TR0                 ;T1中断子程序

        CLR TR1                 ;含16进制→BCD码程序

        CLR C

        MOV A,TL0              ;读T0数值

        RLC A                   ;左移一位即×2

        MOV 60H,A

        MOV A,TH0

        RL A

        MOV ACC.0,C

        MOV 61H,A

        ACALL BCD

        MOV TH0,#00H

        MOV TL0,#00H

        MOV TH1,#3CH

        MOV TL1,#0B0H

        SETB TR0

        SETB TR1

        RETI

         ORG 0400H

BCD:    MOV R7,#05H            ;16进制(61H,60H)→BCD码

         MOV R0,#50H            ;BCD码:54H(万位)-50H(个位)

BCD0:   MOV @R0,#00H

         INC R0

         DJNZ R7,BCD0

HWWEI:  CLR C                ;处理万位

          MOV A,60H

          SUBB A,#10H

          MOV B,A

          MOV A,61H           ;2710H=10000B

          SUBB A,#27H

          JC HQWEI

          INC 54H

          MOV 61H,A

          MOV 60H,B

          AJMP HWWEI

HQWEI:   CLR C                ;处理千位

          MOV A,60H

          SUBB A,#0E8H

          MOV B,A

          MOV A,61H            ;03E8H=1000B

          SUBB A,#03H

          JC HBWEI

          INC 53H

          MOV 61H,A

          MOV 60H,B

          AJMP HQWEI

HBWEI:   CLR C                 ;处理百位

          MOV A,60H

          SUBB A,#64H

          MOV B,A

          MOV A,61H             ;64H=100B

          SUBB A,#00H

          JC HSWEI

          INC 52H

          MOV 61H,A

          MOV 60H,B

          AJMP HBWEI

HSWEI:   CLR C                  ;处理十位

          MOV A,60H

          SUBB A,#0AH

          JC HGWEI               ;0AH=10B

          INC 51H

          MOV 60H,A

          AJMP HSWEI

HGWEI:   MOV 50H,60H            ;处理个位

          MOV 71H,50H            ;修改显示缓冲区

          MOV 72H,51H            ;左移一位即×10

          MOV 73H,52H

          MOV 74H,53H

          MOV 75H,54H

          MOV 70H,#00H

          RET

          ORG 0600H

LED:                                  ;显示子程序

          CLR EA                      ;关中断

          MOV R0,#01H                 ;位码

          MOV R1,#75H                 ;显示缓冲区首址

LOOP:    MOV A,R0                    ;送位码

          MOV DPTR,#6001H

          MOVX @DPTR,A

          MOV A,@R1                  ;送段码

          MOV DPTR,#SEG

          MOVC A,@A+DPTR

          CJNE R0,#04H,LOOP1         ;左边第三位加小数点

DOT:     ADD A,#80H

          MOV DPTR,#6000H

          MOVX @DPTR,A

          ACALL DELAY

          MOV A,#7FH                  ;清黑

          MOVX @DPTR,A

          AJMP GOON

LOOP1:   MOV DPTR,#6000H             ;非左边第三位不加小数点

          MOVX @DPTR,A

          ACALL DELAY

          MOV A,#7FH                   ;清黑

          MOVX @DPTR,A

GOON:    DEC R1

          MOV A,R0                   ;指向下一位

          RL A

          MOV R0,A

          CJNE R0,#40H,LOOP          ;六位显示完返回

          SETB EA                    ;开中断

          RET

   DELAY: MOV R7,#02H

  DELAY0: MOV R6,#0FFH

  DELAY1: DJNZ R6,DELAY1

          DJNZ R7,DELAY0

          RET

     SEG: DB 40H,79H,24H,30H,19H,12H,02H,78H  ;字形表

          DB 00H,10H,08H,03H,46H,21H,06H,0EH

实验二十  温度测量实验

一、目的：

了解温度测量的方法，掌握温度测量系统中软硬件的设计。

二、内容：

利用Pt100热电阻、测量电桥、测量放大器、ICL7135 A/D转换器、8031 

单片机、LED显示器测量并显示水温。

测量显示范围为00.0˚C～99.9˚C。

在程序设计调试过程中，利用电阻箱模拟Pt100热电阻的阻值变化。

三、硬件连接框图见图10-2

    相关电路原理图见第一章图1-2、图1-3、图1-7、图1-8、图1-9。

图 10-2

四、实验步骤：

1、 硬件连接：

1） 电阻箱接入测量电桥输入端Rx；

2） 电阻箱阻值设定为100.0Ω（Pt100热电阻在水温为0˚C时的阻值），

调节电位器WR3（调零电位器）,使测量电桥输出Vout11、Vout12电压为0V；

3） 测量电桥输出Vout11、Vout12接入测量放大器输入端Vin41、Vin42，

调节电位器WR5（调零电位器），使测量放大器输出Vout4电压为0V；

4） 将测量放大器输出Vout4接入ICL7135 A/D转换电路输入端Vin5；

5） ICL7135 A/D转换电路的BCD码数据输出端B8～B1接8255（1）PA口，ICL7135 A/D转换电路的位状态输出端D5~D1（万、千、百、十、个位输出

状态）接8255（1）PB口；

6）  ICL7135 A/D转换电路的转换状态输出端BUSY接8031外部中断输入端，

ICL7135 A/D转换电路的时钟信号输入端CLKIN接入1/8 ALE；

7） 8255（1）片选信号/CS接入/Y3～/Y7（任选其一），

8255（2）片选信号/CS接入/Y3～/Y7（任选其一）；

8） 连接相关硬件模块的电源短路块。

2、 软件设计说明：

A/D转换后将万、千、百、十、个位（BCD码）数据变换成二字节十六进

制数并保存；

六次A/D转换后进行数字滤波，得到A/D转换有效值；

根据A/D转换有效值进行非线性校正及标度变换，得到测量温度值（十六进制数）；

再将此十六进制温度值变换成BCD码（000~999），送去显示。显示时个位数加小数点（00.0˚C ~99.9˚C）。

1） 设计ICL7135 A/D转换程序

2） 确定零点及满量程：

执行A/D转换程序；

调整电阻箱的阻值为100.0Ω，调节电位器WR5（调零电位器）使显示值为00200左右；

调整电阻箱的阻值为138.1Ω，调节电位器WR4（调增益电位器）使显示值为19500左右；

重复调整，达到以上要求。00200及19500为判断欠量程、过量程的阀值。

3） 建立表格：（按下列表格调整电阻箱的阻值，记录显示数据。）

	温度（℃）
	  0．0
	10．0
	20．0
	30．0
	40．0
	50．0

	阻值（Ω）
	 100.0
	103.9
	107.8
	111.7
	115.5
	119.4

	A/D数值
	 
	
	
	
	
	

	十六进制
	
	
	
	
	
	

	
	
	
	
	
	
	

	温度（℃）
	60．0
	70．0
	80．0
	90．0
	99．9
	

	阻值（Ω）
	123.2
	127.1
	130.9
	134.7
	138.1
	

	A/D数值
	
	
	
	
	
	

	十六进制
	
	
	
	
	
	


4） BCD 码→16进制数：

    A/D转换后的数值为BCD码，将其转换成二字节16进制数并保存。

5） 数字滤波程序设计：

当读入A/D转换后数值的次数达到6次时，执行数字滤波程序（去极值求平均值法即去掉最高值及最低值，然后求平均值）；

求得A/D转换的有效值并保存。

6） 设计非线性校正及标度变换程序：

①  建立插值公式：

将上表按温度均匀分成十段，以温度值和A/D采集数据为函数关系建立

十段线性插值公式。

    设直线方程y=ax+b。y是温度值，x是A/D采集数据。

    从表中可以得出每段的（x1,y1）和(x2,y2)。

    求解方程 y1 = ax1+b           则  a = (y2-y1) / (x2-x1)

             y2 = ax2+b               b = (x1y2-x2y1) / (x1-x2)
②  建立插值公式举例：

如果0℃时A/D采集数据为00195D,10℃时A/D采集数据为02300D。

则：  y1=000        x1=195

          y2=100        x2=2300       （y为温度值，包含一位小数）
    根据以上方法求得1/a ≈ 21.1D, b ≈ -1D。
    为了提高测量精度，设m = 1/a×10＝211D＝D3H, n = - b = 1D = 01H。

    则0℃~10℃的插值公式y=ax+b修改为：y=(10×x)/m - n。           
③  根据A/D转换的有效值的大小，选择插值公式，求出被测温度值y。

7） 16进制数→BCD码转换程序设计：
根据插值公式求得的y值为两字节16进制数，将其转换成BCD码（范围

为000～999），再将其送入显示缓冲区。
小数点在确定个位数的字形码时给出,显示范围为00.0˚C～99.9˚C。
8） 当A/D转换的有效值超出欠量程、过量程的阀值时，给出欠量程或过量

程标志（欠量程显示全“－”，过量程显示全“H” ）。

9） 根据以上说明设计程序（在ICL7135 A/D转换程序上修改），将温度值在右边三位LED显示器上显示。

3、 执行程序：

    使用电阻箱模拟Pt100的阻值变化，记录测量结果；

    使用Pt100替换电阻箱，记录测量结果。

*五、高级要求：

查询资料，利用ICL7135与8031单片机二线式连接法（利用A/D转换的时间长短计算出A/D数值），设计新的温度测量系统的软、硬件。将两种方式进行比较。

六、参考程序清单：

参考程序中：

    ICL7135数据输出端B8~B1（BCD码输出）接8255（1）PA口的PA7~PA4，

ICL7135位状态输出端D4~D1（千、百、十、个位输出状态）接PA3~PA0；

    判断万位时，无千、百、十、个位输出时，即为万位输出；

    ICL7135的时钟输入端CLKIN接1/8 ALE，转换状态输出端BUSY接8031

的外部中断输入端 /INT0；

8255（1）片选信号 /CS接 /Y7；

8255（2）片选信号 /CS接 /Y3。

表格记录如下：

	温度（℃）
	  0.0
	10.0
	20.0
	30.0
	40.0
	50.0

	阻值（Ω）
	 100.0
	103.9
	107.8
	111.7
	115.5
	119.4

	A/D数值
	 00195
	02300
	04319
	06355
	0302
	10285

	 十六进制
	00C3H
	08FCH
	10DFH
	18D3H
	206EH
	282DH

	
	
	
	
	
	
	

	温度（℃）
	60.0
	70.0
	80.0
	90.0
	99．9
	

	阻值（Ω）
	123.2
	127.1
	130.9
	134.7
	138.1
	

	A/D数值
	12256
	14201
	16161
	18025
	19627
	

	十六进制
	2FE0H
	3779H
	3F21H
	4669H
	4CABH
	


          ORG 0000H

          AJMP START

          ORG 0003H              ;/INT0中断矢量地址入口

          LJMP INTER

          ORG 0100H

   START: MOV R0,#50H

          MOV R2,#05H

    PUT1: MOV @R0,#00H

          INC R0                 ;数据区赋值全0(54H-50H)

          DJNZ R2,PUT1

          MOV R0,#56H

          MOV R2,#03H

    PUT2: MOV @R0,#10H           ;给显示缓冲区(58H-56H)赋值显示全黑

          INC R0

          DJNZ R2,PUT2

          MOV DPTR,#6003H

          MOV A,#81H

          MOVX @DPTR,A           ;对8255(2)初始化(接LED显示器)

          MOV A,#92H             ;对8255(1)初始化(接ICL7135)

          MOV DPTR,#0E003H

          MOVX @DPTR,A

          MOV SP,#70H            ;设置堆栈区

          MOV IE,#00H            ;清中断

          SETB IT0               ;设置跳变触发

          SETB EA

          SETB EX0               ;开中断

    MAIN: ACALL LED              ;调显示子程序

          NOP

          NOP

          AJMP MAIN

     LED: CLR EA                  ;显示子程序，关中断

          MOV R0,#20H

          MOV R1,#56H

    LOOP: MOV A,R0                ;送位码(PB口)

          MOV DPTR,#6001H

          MOVX @DPTR,A

          MOV A,@R1               ;送段码(PA口)

          MOV DPTR,#SEG

          MOVC A,@A+DPTR

          CJNE R0,#10H,LOOP1

     DOT: ADD A,#80H

          MOV DPTR,#6000H        ;个位带小数点

          MOVX @DPTR,A

          AJMP GOON

   LOOP1: MOV DPTR,#6000H        ;非个位不带小数点

          MOVX @DPTR,A

    GOON: ACALL DELAY

          MOV A,#7FH

          MOVX @DPTR,A            ;清黑

          MOV A,R0

          RR A

          MOV R0,A

          INC R1

          CJNE R0,#04H,LOOP       ;三位显示完？

          SETB EA

          RET

   DELAY: MOV R7,#02H

  DELAY0: MOV R6,#0FFH

  DELAY1: DJNZ R6,DELAY1

          DJNZ R7,DELAY0

          RET

          ORG 1000H

   INTER: PUSH PSW         ;中断服务子程序(现场保护)

          PUSH DPH

          PUSH DPL

          PUSH ACC

          CLR EA           ;关中断

     D5:  MOV DPTR,#0E000H

          MOVX A,@DPTR     ;读PA口(PA3-PA0),如全为低,表明是万位

          MOV R3,A         ;否则等待

          ANL A,#0FH

          JNZ D5

          MOV A,R3

          ANL A,#0F0H      ;读万位BCD码(PA7-PA4)

          SWAP A

          MOV 54H,A        ;放入地址54H(万位)

     D4:  MOV DPTR,#0E000H

          MOVX A,@DPTR

          MOV R3,A

          ANL A,#08H       ;D4为高表明上千位

          JZ D4            ;否则等待

          MOV A,R3

          ANL A,#0F0H

          SWAP A

          MOV 53H,A        ;放入地址53H(千位)

     D3:  MOV DPTR,#0E000H

          MOVX A,@DPTR

          MOV R3,A

          ANL A,#04H

          JZ D3

          MOV A,R3

          ANL A,#0F0H

          SWAP A

          MOV 52H,A         ;52H(百位)

     D2:  MOV DPTR,#0E000H

          MOVX A,@DPTR

          MOV R3,A

          ANL A,#02H

          JZ D2

          MOV A,R3

          ANL A,#0F0H

          SWAP A

          MOV 51H,A         ;51H(十位)

     D1:  MOV DPTR,#0E000H

          MOVX A,@DPTR

          MOV R3,A

          ANL A,#01H

          JZ D1

          MOV A,R3

          ANL A,#0F0H

          SWAP A

          MOV 50H,A          ;50H(个位)

  RETURN: ACALL BTOH         ;调(BCD码→16进制)子程序

          ACALL CHANGE       ;调(非线性校正及标度变换)子程序

          POP ACC            ;恢复现场

          POP DPL

          POP DPH

          POP PSW

          SETB EA            ;开中断

          RETI

    BTOH: MOV 60H,50H    ;BCD码(54H-50H)→16进制(61H,60H)

          MOV 61H,#00H   ;个位

   BSWEI: MOV A,51H          ;十位

          JZ BBWEI

          MOV R7,51H

      SS: MOV A,60H

          ADD A,#0AH         ;0AH=10B

          MOV 60H,A

          MOV A,61H

          ADDC A,#00H

          MOV 61H,A

          DJNZ R7,SS

   BBWEI: MOV A,52H           ;百位

          JZ BQWEI

          MOV R7,52H

      BB: MOV A,60H

          ADD A,#64H           ;64H=100B

          MOV 60H,A

          MOV A,61H

          ADDC A,#00H

          MOV 61H,A

          DJNZ R7,BB

   BQWEI: MOV A,53H            ;千位

          JZ BWWEI

          MOV R7,53H

      QQ: MOV A,60H

          ADD A,#0E8H

          MOV 60H,A

          MOV A,61H            ;03E8H=1000B

          ADDC A,#03H

          MOV 61H,A

          DJNZ R7,QQ

   BWWEI: MOV A,54H             ;万位

          JZ QU1

          MOV R7,54H

      WW: MOV A,60H

          ADD A,#10H

          MOV 60H,A

          MOV A,61H              ;2710H=10000B

          ADDC A,#27H

          MOV 61H,A

          DJNZ R7,WW

     QU1: RET

  CHANGE: MOV R0,60H ;非线性校及标度变换(含16进制→BCD码)

          MOV R1,61H ;按温度分段

          CJNE R1,#00H,L1   ;与0℃(00C3H)比较。不等转至L1

          CJNE R0,#0C3H,L2  ;                  不等转至L2

    CHA1: MOV A,#00H        ;0℃-10℃

          AJMP GO

      L1: CJNE R1,#08H,L3   ;与10℃(08FCH)比较。不等转至L3

          CJNE R0,#0FCH,L4  ;                   不等转至L4

    CHA2: MOV A,#02H        ;10℃-20℃

          AJMP GO

      L2: JC ERROR1         ;即小于0℃时转 ERROR1

          AJMP CHA1         ;即高位为#00H,必在0℃-10℃之间

  ERROR1: MOV 56H,#11H      ;三位全部显示“-”

          MOV 57H,#11H

          MOV 58H,#11H

          AJMP QU2

      L3: JC CHA1           ;C=1,(61H)<#08H,必在0℃-10℃

          CJNE R1,#10H,L5   ;与20℃(10DFH)比较。

          CJNE R0,#0DFH,L6

    CHA3: MOV A,#04H        ;20℃-30℃

          AJMP GO

      L4: JC CHA1           ;C=1,(61H)=08H,(60H)<FCH,必在0℃-10℃

          AJMP CHA2         ;C=0,(61H)=08H,(60H)>FCH,必在10℃-20℃

      L5: JC CHA2               ;C=1,(61H)<10H,必在10℃-20℃

          CJNE R1,#18H,L7       ;与30℃(18D3H)比较。

          CJNE R0,#0D3H,L8

    CHA4: MOV A,#06H            ;30℃-40℃

          AJMP GO

      L6: JC CHA2         ;C=1,(61H)=10H,(60H)<DFH,必在10℃-20℃

          AJMP CHA3       ;C=0,(61H)=10H,(60H)>DFH,必在20℃-30℃

      L7: JC CHA3

          CJNE R1,#20H,L9       ;与40℃(206EH)比较。

          CJNE R0,#6EH,L10

    CHA5: MOV A,#08H            ;40℃-50℃

          AJMP GO

      L8: JC CHA3

          AJMP CHA4

      L9: JC CHA4

          CJNE R1,#28H,L11      ;与50℃(282DH)比较。

          CJNE R0,#2DH,L12

    CHA6: MOV A,#0AH            ;50℃-60℃

          AJMP GO

     L10: JC CHA4

          AJMP CHA5

     L11: JC CHA5

          CJNE R1,#2FH,L13      ;与60℃(2FE0H)比较。

          CJNE R0,#0E0H,L14

    CHA7: MOV A,#0CH            ;60℃-70℃

          AJMP GO

     L12: JC CHA5

          AJMP CHA6

     L13: JC CHA6

          CJNE R1,#37H,L15      ;与70℃(3779H)比较。

          CJNE R0,#79H,L16

    CHA8: MOV A,#0EH            ;70℃-80℃

          AJMP GO

     L14: JC CHA6

          AJMP CHA7

     L15: JC CHA7

          CJNE R1,#3FH,L17      ;与80℃(3F21H)比较。

          CJNE R0,#21H,L18

    CHA9: MOV A,#10H            ;80℃-90℃

          AJMP GO

     L16: JC CHA7

          AJMP CHA8

     L17: JC CHA8

          CJNE R1,#46H,L19      ;与90℃(4669H)比较。

          CJNE R0,#69H,L20

   CHA10: MOV A,#12H            ;90℃-99.9℃

          AJMP GO

     L18: JC CHA8

          AJMP CHA9

     L19: JC CHA9

          CJNE R1,#4CH,L21     ;与99.9℃(4CABH)比较。

          CJNE R0,#0ABH,L22

          AJMP CHA10           ;恰好为99.9℃时转CHA10

     L20: JC CHA9

          AJMP CHA10

     L21: JC CHA10

          AJMP ERROR2

L22: JC CHA10

  ERROR2: MOV 56H,#12H          ;三位全部显示“H”

          MOV 57H,#12H

          MOV 58H,#12H

          AJMP QU2

      GO: MOV R5,A        ;计算y=(10×x)/m - n

          MOV DPTR,#TABLE

          MOVC A,@A+DPTR

          MOV 37H,A       ;查表将m放入37H(除数)

          MOV A,R5

          INC A

          MOVC A,@A+DPTR

          MOV 38H,A       ;查表将n放入38H

  MULKLM: CLR C

          MOV A,#0AH      ;计算 y=10×x

          MOV B,60H       ;X为(61H,60H)

          MUL AB

          MOV 67H,A       ;积为(69H,68,67H)

          MOV 68H,B

          MOV A,#0AH

          MOV B,61H

          MUL AB

          ADDC A,68H

          MOV 68H,A

          MOV A,B

          ADDC A,#00H

          MOV 69H,A

          MOV 43H,#00H    ;计算(10×x)/m

          MOV 42H,#00H    ;被除数(69H,68H,67H) 除数(37H)

          MOV 62H,#00H    ;商(43H,42H) 余数(62H)

    DIVD: CLR C

          MOV A,67H

          SUBB A,37H

          MOV B,A

          JC QUIT1       ;不够减的话转至 QUIT1

          MOV 67H,A

          AJMP G         ;够减的话转至 G

   QUIT1: MOV A,68H

          JZ QUIT3  ;最低位不够减且第二位为0,向最高位借位转QUIT3

   QUIT2: MOV 67H,B ;最低位不够减但第二位不为0时

          MOV A,68H

          SUBB A,#00H

          MOV 68H,A

          AJMP G     ;此部分处理完毕后转至 G

   QUIT3: MOV A,69H

          JZ SUBB1   ;借位但最高位和第二位都为0时除法转换结束

          SUBB A,#00H

          MOV 69H,A

          MOV 68H,#0FFH

          MOV A,#0FFH

          SUBB A,37H

          ADD A,67H

          INC A

          MOV 67H,A

       G: MOV 62H,67H        ;62H放余数

          MOV A,42H

          ADD A,#01H

          MOV 42H,A

          MOV A,43H

          ADDC A,#00H

          MOV 43H,A

          AJMP DIVD

   SUBB1: MOV A,62H          ;处理余数

          CJNE A,#5DH,SUBB2  ;不等转SUBB2,进行下一步判别

          JC SUBB3           ;相等(5入),跳转至SUBB3

   SUBB2: JC SUBB4           ;(62H)<5DH,(4舍)并跳转至SUBB4

   SUBB3: CLR C

          MOV A,42H

          ADD A,#01H

          MOV 42H,A

          MOV A,43H

          ADDC A,#00H

          MOV 43H,A

   SUBB4: CLR C              ;计算 (10×x/m) - n

          MOV A,42H

          SUBB A,38H

          MOV 42H,A

          MOV A,43H

          SUBB A,#00H

          MOV 43H,A

    HTOB: MOV 56H,#00H        ;16进制(43H,42H)→BCD码

          MOV 57H,#00H        ;BCD码:5AH(千位)-56H(小数位)

          MOV 58H,#00H        ;测量值范围(00.0-99.9)

          MOV 59H,#00H        ;实际存放(58H,57H,56H)

          MOV 5AH,#00H

   HWWEI: CLR C               ;处理千位

          MOV A,42H

          SUBB A,#10H

          MOV B,A

          MOV A,43H           ;2710H=10000B(含小数位故为五位数)

          SUBB A,#27H

          JC HQWEI

          INC 5AH

          MOV 43H,A

          MOV 42H,B

          AJMP HWWEI

   HQWEI: CLR C                ;处理百位

          MOV A,42H

          SUBB A,#0E8H

          MOV B,A

          MOV A,43H            ;03E8H=1000B

          SUBB A,#03H

          JC HBWEI

          INC 59H

          MOV 43H,A

          MOV 42H,B

          AJMP HQWEI

   HBWEI: CLR C                 ;处理十位

          MOV A,42H

          SUBB A,#64H

          MOV B,A

          MOV A,43H             ;64H=100B

          SUBB A,#00H

          JC HSWEI

          INC 58H

          MOV 43H,A

          MOV 42H,B

          AJMP HBWEI

   HSWEI: CLR C                  ;处理个位

          MOV A,42H

          SUBB A,#0AH

          JC HGWEI               ;0AH=10B

          INC 57H

          MOV 42H,A

          AJMP HSWEI

   HGWEI: MOV 56H,42H            ;处理小数位

     QU2: RET

     SEG: DB 40H,79H,24H,30H,19H,12H,02H,78H

          DB 00H,10H,08H,03H,46H,21H,06H,0EH

          DB 7FH,3FH,09H                            ;段码

   TABLE: DB 0D3H,09H,0CAH,09H,0CCH,0CH,0C3H,1AH

          DB 0C6H,13H,0C5H,16H,0C3H,1EH,0C4H,19H

          DB 0BAH,43H,0A2H,0D6H                     ;参数m,n
实验二十一  温度测量控制实验
一、实验目的：

了解温度测量控制的方法，掌握温度测量控制系统中软硬件的设计。

二、实验内容：

利用Pt100热电阻、测量电桥、测量放大器、ICL7135 A/D转换器、8031 

单片机测量水温；

利用固态继电器、电热杯控制加热水温；

利用键盘设定温度并控制系统工作；

LED显示器左边三位显示设定水温，右边三位显示测量水温；

测量、控制、显示水温范围为00.0˚C～99.9˚C；

在程序设计调试过程中，利用电阻箱模拟Pt100热电阻的阻值变化。
三、硬件连接框图见图10-3。

相关电路原理图见第一章图1-2、图1-3、图1-7、图1-8、图1-9、图1-11。

图10-3

四、 实验步骤：

1、 硬件连接：

1） 电阻箱接入测量电桥输入端Rx，电阻箱阻值设定为100.0Ω（Pt100热
电阻在水温为0˚C时的阻值）；

调节电位器WR3,使测量电桥输出Vout11、Vout12电压为0V。

2） 测量电桥输出Vout11、Vout12接入测量放大器输入端Vin41、Vin42；

调节电位器WR5，使测量放大器输出Vout4电压为0V。

3） 将测量放大器输出Vout4接入ICL7135 A/D转换电路输入端Vin5。

4） ICL7135 A/D转换电路的BCD码数据输出端B8～B1接8255（1）PA口；

ICL7135 A/D的位状态输出端D5~D1（万、千、百、十、个位输出状态）接8255（1）PB口；

ICL7135 A/D转换状态输出端BUSY接8031外部中断输入端；

ICL7135 A/D转换电路的时钟信号输入端CLKIN接入1/8 ALE。

5） 控制通道：

8255（1）PC口的PC0接CONin；

CONout(+/-) 接固态继电器控制信号输入端(+/-) ；

固态继电器控制信号输出端接电热杯；

固态继电器电源端接交流220V；

电热杯内必须加水！

6） 8255（1）片选信号/CS接入/Y3～/Y7（任选其一）；

    8255（2）片选信号/CS接入/Y3～/Y7（任选其一）。    

7） 连接相关硬件模块的电源短路块。

2、 软件设计说明：

利用键盘设定温度并控制系统工作；

LED显示器左边三位显示设定温度，右边三位显示测量温度；

A/D转换后将万、千、百、十、个位（BCD码）数据变换成二字节十六进

制数并保存；

六次A/D转换后进行数字滤波，得到A/D转换有效值；

根据A/D转换有效值进行非线性校正及标度变换，得到测量温度值（十六进制数）；

再将此十六进制温度值变换成BCD码（000~999），送去显示，显示时个位数加小数点（00.0˚C ~99.9˚C）；

    将设定值与测量值进行比较，确定控制通道的工作。

1） 设计ICL7135 A/D转换程序

2） 确定零点及满量程：

  执行A/D转换程序；
  调整电阻箱的阻值为100.0Ω，调节电位器WR5（调零电位器）使显示值为00200左右；

  调整电阻箱的阻值为138.1Ω，调节电位器WR4（调增益电位器）使显示值为19500左右；

  重复调整，达到以上要求；

  00200及19500为判断欠量程、过量程的阀值。

3） 建立表格：（按下列表格调整电阻箱的阻值，记录显示数据。）

	温度（℃）
	  0．0
	10．0
	20．0
	30．0
	40．0
	50．0

	阻值（Ω）
	 100.0
	103.9
	107.8
	111.7
	115.5
	119.4

	A/D数值
	 
	
	
	
	
	

	十六进制
	
	
	
	
	
	

	
	
	
	
	
	
	

	温度（℃）
	60．0
	70．0
	80．0
	90．0
	99．9
	

	阻值（Ω）
	123.2
	127.1
	130.9
	134.7
	138.1
	

	A/D数值
	
	
	
	
	
	

	十六进制
	
	
	
	
	
	


4） BCD 码→16进制数：

    读入A/D转换后的数值为BCD码，将其转换成二字节16进制数并保存。

5） 数字滤波程序设计：

当读入A/D转换后的数值次数达到6次时，执行数字滤波程序（去极值求平均值法即去掉最高值及最低值，然后求平均值）；

求得A/D转换的有效值并保存。

6） 设计非线性校正及标度变换程序：

①  建立插值公式：

将上表按温度均匀分成十段，以温度值和A/D采集数据为函数关系建立十段线性插值公式。

    设直线方程y=ax+b。y是温度值，x是A/D采集数据。

    从表中可以得出每段的（x1,y1）和(x2,y2)。

    求解方程 y1=ax1+b           则 a=(y2-y1)/(x2-x1)

             y2=ax2+b                 b= (x1y2-x2y1)/(x1-x2)
②  建立插值公式举例：

如果0℃时A/D采集数据为00195D,10℃时A/D采集数据为02300D。

    则：  y1=000        x1=195

          y2=100        x2=2300           （y为温度值，包含一位小数）
    根据以上方法求得1/a ≈ 21.1D, b ≈ -1D。
    为了提高测量精度，设m = 1/a×10＝211D＝D3H, n = - b = 1D = 01H。

    则0℃—10℃的插值公式y=ax+b修改为：y=(10×x)/m - n。           
③  根据A/D转换的有效值的大小，选择插值公式，求出被测温度值y。
7） 16进制数→BCD码转换程序设计：
根据插值公式求得的y值为两字节16进制数，将其转换成BCD码（范围

为000～999）并保存。
8） 键盘设定温度并控制系统工作：

    数字键为0~9。

A键为工作键，B键为暂停（设定）键。

9） 根据以上要求设计程序（修改ICL7135 A/D转换程序）；

测量温度值在右边三位LED显示器上显示；

设定温度值在左边三位LED显示器上显示；

显示范围均为00.0˚C～99.9˚C。小数点在确定个位数的字形码时给出；

    当A/D转换的有效值超出欠量程、过量程的阀值时，给出欠量程或过量程

标志（欠量程显示全“－”，过量程显示全“H” ）。

10）温度控制方式：

    采用脉宽调制方式。

当（设定键─测量值）≧ 6˚C 时，全加热；

当（设定键─测量值）在6˚C ~ 4˚C 时，加热通/断时间比5:1 ；         

当（设定键─测量值）在4˚C ~ 2˚C 时，加热通/断时间比1:1 ；

当（设定键─测量值）在2˚C ~ 0˚C 时，加热通/断时间比1:5 ；

当（设定键─测量值）≦0˚C 时，全关断。

3、 根据以上要求设计（修改）程序；

执行程序；

使用电阻箱模拟Pt100的阻值变化，记录测量控制结果；

使用Pt100替换电阻箱，将Pt100插入电热杯中，记录测量控制结果。
*五、高级要求：
了解其它形式的控制方式，设计更精确的温度控制程序。

5、 参考程序：
参考程序中：
数字键为0~9；

A键为工作键，B键为暂停（设定）键；
    ICL7135数据输出端B8~B1（BCD码输出）接8255（1）PA口的PA7~PA4；

ICL7135位状态输出端D4~D1（千、百、十、个位输出状态）接PA3~PA0；

    判断万位时，无千、百、十、个位输出时，即为万位输出。

ICL7135的时钟输入端CLKIN接1/8 ALE；

转换状态输出端BUSY接8031的外部中断输入端 /INT0；

8255（1）PC口的PC0接CONin,；

CONout(+/-) 接固态继电器控制信号输入端(+/-) ；

固态继电器控制信号输出端接电热杯；

固态继电器电源端接交流220V；

电热杯内必须加水！

温度控制采用脉宽调制方式：

当（设定键─测量值）≧ 6˚C 时，全加热；

    当（设定键─测量值）在6˚C ~ 4˚C 时，加热通/断时间比5:1 ；         

    当（设定键─测量值）在4˚C ~ 2˚C 时，加热通/断时间比1:1 ；

    当（设定键─测量值）在2˚C ~ 0˚C 时，加热通/断时间比1:5 ；

    当（设定键─测量值）≦0˚C 时，全关断。

8255（1）片选信号 /CS接 /Y7；

8255（2）片选信号 /CS接 /Y3。

    记录的数据如下：
	温度（℃）
	  0.0
	10.0
	20.0
	30.0
	40.0
	50.0

	阻值（Ω）
	 100.0
	103.9
	107.8
	111.7
	115.5
	119.4

	A/D数值
	 00195
	02300
	04319
	06355
	0302
	10285

	 十六进制
	00C3H
	08FCH
	10DFH
	18D3H
	206EH
	282DH

	
	
	
	
	
	
	

	温度（℃）
	60.0
	70.0
	80.0
	90.0
	99．9
	

	阻值（Ω）
	123.2
	127.1
	130.9
	134.7
	138.1
	

	A/D数值
	12256
	14201
	16161
	18025
	19627
	

	十六进制
	2FE0H
	3779H
	3F21H
	4669H
	4CABH
	


          ORG 0000H

          AJMP START

          ORG 0003H       ;/INT0中断矢量地址入口

          LJMP INTER

          ORG 0100H

   START: MOV R0,#50H

          MOV R2,#05H

    PUT1: MOV @R0,#00H

          INC R0        ;A/D值(BCD码)存储单元(54H-50H)清0

          DJNZ R2,PUT1

          MOV R0,#20H

          MOV R2,#0CH

    PUT2: MOV @R0,#00H  ;六组A/D值(16进制)存储单元(2BH-20H)清0

          INC R0

          DJNZ R2,PUT2

          MOV R0,#56H

          MOV R2,#03H

    PUT3: MOV @R0,#00H  ;测量值显示缓冲区(58H-56H)清0

          INC R0

          DJNZ R2,PUT3

          MOV R0,#4DH

          MOV R2,#03H

    PUT4: MOV @R0,#00H  ;设定值显示缓冲区(4FH-4DH)清0

          INC R0

          DJNZ R2,PUT4

          MOV 2CH,#00H  ;记录A/D转换次数

          MOV 2DH,#20H  ;六组A/D值(16进制)存储单元(2BH-20H)指针

          MOV 2EH,#01H  ;加热控制系数

          MOV A,#92H

          MOV DPTR,#0E003H

          MOVX @DPTR,A      ;8255(1)初始化(接ICL7135)

          MOV DPTR,#6003H

          MOV A,#81H

          MOVX @DPTR,A      ;8255(2)初始化(显示器键盘)

          MOV DPTR,#0E002H

          MOV A,#00H

          MOVX @DPTR,A      ;关加热

          MOV SP,#70H       ;设置堆栈区

          CLR F0            ;F0作温度值设定标志

          SETB IT0          ;设置跳变触发

          SETB EX0

    MAIN: LCALL LED         ;调显示子程序

          SETB EA

          NOP

          NOP

          NOP

          NOP

          CLR EA

          MOV A,2CH       ;A/D采集次数为6次？

          CJNE A,#06H,H0  ;采集6次后,数据处理

          MOV 2CH,#00H    ;A/D采集次数

          MOV 2DH,#20H    ;六组A/D值(16进制)存储单元(2BH-20H)指针

          LCALL MAX         ;去最大值

          LCALL MIN         ;去最小值

          LCALL JIA         ;求平均值

          MOV 60H,22H       ;(23H,22H)为平均值(16进制)

          MOV 61H,23H

          LCALL CHANGE      ;非线性校正及标度变换

          JNB F0,H0

          LCALL CONTROL     ;设定值已设定则调控制子程序

          MOV A,2EH         ;加热控制系数

          CJNE A,#06H,MAIN1

          MOV A,#01H

          MOV 2EH,A

          AJMP H0

   MAIN1: INC A

          MOV 2EH,A

      H0: LCALL CHAJIAN     ;调键盘子程序

          MOV A,R2

          CJNE A,#0BH,MAIN  ;非暂停键(B键)返回

          MOV DPTR,#0E002H  ;关加热

          MOV A,#00H

          MOVX @DPTR,A

      H1: MOV R0,#4FH       ;设定值显示缓冲区首地址

      H2: LCALL LED

          LCALL CHAJIAN     ;调键盘子程序

          MOV A,R2

          CJNE A,#0FFH,H3

          AJMP H2           ;无键按下等待

      H3: MOV A,R2

          CJNE A,#0AH,H4    ;是工作键(A键)返回

          SETB F0           ;表示设定值设定完毕

          AJMP MAIN

      H4: JC H5             ;是数字键(小于A)去处理设定值

          AJMP H2           ;非数字键等待

      H5: MOV A,R2

          MOV @R0,A         ;处理设定值

          DEC R0

          CJNE R0,#04CH,H2

          AJMP H1

     LED: CLR EA            ;显示子程序

          PUSH PSW

          SETB RS0

          CLR RS1           ;一区工作寄存器

          MOV DPTR,#6003H   ;8255(2)初始化(接显示器键盘)

          MOV A,#81H

          MOVX @DPTR,A

          MOV R0,#01H       ;左三位显示设定值

          MOV R1,#4FH

    LOOP: MOV A,R0              ;送位码(B口)

          MOV DPTR,#6001H

          MOVX @DPTR,A

          MOV A,@R1             ;送段码(A口)

          MOV DPTR,#SEG

          MOVC A,@A+DPTR

          CJNE R0,#02H,LOOP1

          ADD A,#80H

          MOV DPTR,#6000H       ;个位数带小数点

          MOVX @DPTR,A

          AJMP GOON

   LOOP1: MOV DPTR,#6000H

          MOVX @DPTR,A

    GOON: LCALL DELAY         ;延时

          MOV A,#7FH          ;清显示

          MOVX @DPTR,A

          MOV A,R0

          RL A

          MOV R0,A

          DEC R1

          CJNE R0,#08H,LOOP

  DISPLAY:MOV R0,#08H          ;右三位显示测量值

          MOV R1,#58H

     LP0: MOV A,R0

          MOV DPTR,#6001H

          MOVX @DPTR,A        ;送位码(B口)

          MOV A,@R1           ;送段码(A口)

          MOV DPTR,#SEG

          MOVC A,@A+DPTR

          CJNE R0,#10H,LP1

          ADD A,#80H

          MOV DPTR,#6000H     ;个位数带小数点

          MOVX @DPTR,A

          AJMP LP2

     LP1: MOV DPTR,#6000H

          MOVX @DPTR,A

     LP2: LCALL DELAY

          MOV A,#7FH          ;清显示

          MOVX @DPTR,A

          MOV A,R0

          RL A

          MOV R0,A

          DEC R1

          CJNE R0,#40H,LP0

          POP PSW

          RET

     MAX: CLR C               ;(2BH,2AH)-(21H,20H)最大值放入(21H,20H)

          MOV A,20H

          SUBB A,22H

          MOV A,21H

          SUBB A,23H

          JC APUT1

          LJMP AXX1

   APUT1: MOV A,20H

          MOV 20H,22H

          MOV 22H,A

          MOV B,21H

          MOV 21H,23H

          MOV 23H,B

    AXX1: CLR C

          MOV A,20H

          SUBB A,24H

          MOV A,21H

          SUBB A,25H

          JC APUT2

          LJMP AXX2

   APUT2: MOV A,20H

          MOV 20H,24H

          MOV 24H,A

          MOV B,21H

          MOV 21H,25H

          MOV 25H,B

    AXX2: CLR C

          MOV A,20H

          SUBB A,26H

          MOV A,21H

          SUBB A,27H

          JC APUT3

          LJMP AXX3

   APUT3: MOV A,20H

          MOV 20H,26H

          MOV 26H,A

          MOV B,21H

          MOV 31H,27H

          MOV 27H,B

    AXX3: CLR C

          MOV A,20H

          SUBB A,28H

          MOV A,21H

          SUBB A,29H

          JC APUT4

          LJMP AXX4

   APUT4: MOV A,20H

          MOV 20H,28H

          MOV 28H,A

          MOV B,21H

          MOV 21H,29H

          MOV 29H,B

    AXX4: CLR C

          MOV A,20H

          SUBB A,2AH

          MOV A,21H

          SUBB A,2BH

          JC APUT5

          LJMP AXX5

   APUT5: MOV A,20H

          MOV 20H,2AH

          MOV 2AH,A

          MOV B,21H

          MOV 21H,2BH

          MOV 2BH,B

    AXX5: RET

     MIN: CLR C           ;(2BH,2AH)-(23H,22H)最小值放入(2BH,2AH)

          MOV A,2AH

          SUBB A,22H

          MOV A,2BH

          SUBB A,23H

          JNC PUT11

          AJMP XX1

   PUT11: MOV A,2AH

          MOV 2AH,22H

          MOV 2AH,A

          MOV B,2BH

          MOV 2BH,23H

          MOV 23H,B

     XX1: CLR C

          MOV A,2AH

          SUBB A,24H

          MOV A,2BH

          SUBB A,25H

          JNC PUT22

          AJMP XX2

   PUT22: MOV A,2AH

          MOV 2AH,24H

          MOV 24H,A

          MOV B,2BH

          MOV 2BH,25H

          MOV 25H,B

     XX2: CLR C

          MOV A,2AH

          SUBB A,26H

          MOV A,2BH

          SUBB A,27H

          JNC PUT33

          AJMP XX3

   PUT33: MOV A,2AH

          MOV 2AH,26H

          MOV 26H,A

          MOV B,2BH

          MOV 2BH,27H

          MOV 27H,B

     XX3: CLR C

          MOV A,2AH

          SUBB A,28H

          MOV A,29H

          JNC PUT44

          AJMP XX4

   PUT44: MOV A,2AH

          MOV 2AH,28H

          MOV 28H,A

          MOV B,2BH

          MOV 2BH,29H

          MOV 29H,B

     XX4: RET

     JIA: MOV A,22H  ;(29H,28H)-(23H,22H)平均值放入(23H,22H)

          ADD A,24H

          MOV 22H,A

          MOV A,23H

          ADDC A,25H

          MOV 23H,A

          MOV A,22H

          ADD A,26H

          MOV 22H,A

          MOV A,23H

          ADDC A,27H

          MOV 23H,A

          MOV A,22H

          ADD A,28H

          MOV 22H,A

          MOV A,23H

          ADDC A,29H

          MOV 23H,A

          MOV A,23H

          RRC A

          MOV 23H,A

          MOV A,22H

          RRC A

          MOV 22H,A

          CLR C

          MOV A,23H

          RRC A

          MOV 23H,A

          MOV A,22H

          RRC A

          MOV 22H,A

          RET

  CHANGE: MOV R0,60H ;非线性校及标度变换(含16进制→BCD码)

          MOV R1,61H ;按温度分段

          CJNE R1,#00H,L1   ;与0℃(00C3H)比较。不等转至L1

          CJNE R0,#0C3H,L2  ;                  不等转至L2

    CHA1: MOV A,#00H        ;0℃-10℃

          AJMP GO

      L1: CJNE R1,#08H,L3   ;与10℃(08FCH)比较。不等转至L3

          CJNE R0,#0FCH,L4  ;                   不等转至L4

    CHA2: MOV A,#02H        ;10℃-20℃

          AJMP GO

      L2: JC ERROR1         ;即小于0℃时转 ERROR1

          AJMP CHA1         ;即高位为#00H,必在0℃-10℃之间

  ERROR1: MOV 56H,#11H      ;三位全部显示“-”

          MOV 57H,#11H

          MOV 58H,#11H

          AJMP QU2

      L3: JC CHA1           ;C=1,(61H)<#08H,必在0℃-10℃

          CJNE R1,#10H,L5   ;与20℃(10DFH)比较。

          CJNE R0,#0DFH,L6

    CHA3: MOV A,#04H        ;20℃-30℃

          AJMP GO

      L4: JC CHA1           ;C=1,(61H)=08H,(60H)<FCH,必在0℃-10℃

          AJMP CHA2         ;C=0,(61H)=08H,(60H)>FCH,必在10℃-20℃

      L5: JC CHA2               ;C=1,(61H)<10H,必在10℃-20℃

          CJNE R1,#18H,L7       ;与30℃(18D3H)比较。

          CJNE R0,#0D3H,L8

    CHA4: MOV A,#06H            ;30℃-40℃

          AJMP GO

      L6: JC CHA2               ;C=1,(61H)=10H,(60H)<DFH,必在10℃-20℃

          AJMP CHA3             ;C=0,(61H)=10H,(60H)>DFH,必在20℃-30℃

      L7: JC CHA3

          CJNE R1,#20H,L9       ;与40℃(206EH)比较。

          CJNE R0,#6EH,L10

    CHA5: MOV A,#08H            ;40℃-50℃

          AJMP GO

      L8: JC CHA3

          AJMP CHA4

      L9: JC CHA4

          CJNE R1,#28H,L11      ;与50℃(282DH)比较。

          CJNE R0,#2DH,L12

    CHA6: MOV A,#0AH            ;50℃-60℃

          AJMP GO

     L10: JC CHA4

          AJMP CHA5

     L11: JC CHA5

          CJNE R1,#2FH,L13      ;与60℃(2FE0H)比较。

          CJNE R0,#0E0H,L14

    CHA7: MOV A,#0CH            ;60℃-70℃

          AJMP GO

     L12: JC CHA5

          AJMP CHA6

     L13: JC CHA6

          CJNE R1,#37H,L15      ;与70℃(3779H)比较。

          CJNE R0,#79H,L16

    CHA8: MOV A,#0EH            ;70℃-80℃

          AJMP GO

     L14: JC CHA6

          AJMP CHA7

     L15: JC CHA7

          CJNE R1,#3FH,L17      ;与80℃(3F21H)比较。

          CJNE R0,#21H,L18

    CHA9: MOV A,#10H            ;80℃-90℃

          AJMP GO

     L16: JC CHA7

          AJMP CHA8

     L17: JC CHA8

          CJNE R1,#46H,L19      ;与90℃(4669H)比较。

          CJNE R0,#69H,L20

   CHA10: MOV A,#12H            ;90℃-99.9℃

          AJMP GO

     L18: JC CHA8

          AJMP CHA9

     L19: JC CHA9

          CJNE R1,#4CH,L21     ;与99.9℃(4CABH)比较。

          CJNE R0,#0ABH,L22

          AJMP CHA10           ;恰好为99.9℃时转CHA10

     L20: JC CHA9

          AJMP CHA10

     L21: JC CHA10

          AJMP ERROR2

     L22: JC CHA10

  ERROR2: MOV 56H,#12H          ;三位全部显示“H”

          MOV 57H,#12H

          MOV 58H,#12H

          AJMP QU2

      GO: MOV R5,A        ;计算y=(10×x)/m - n

          MOV DPTR,#TABLE

          MOVC A,@A+DPTR

          MOV 37H,A       ;查表将m放入37H(除数)

          MOV A,R5

          INC A

          MOVC A,@A+DPTR

          MOV 38H,A       ;查表将n放入38H

  MULKLM: CLR C

          MOV A,#0AH      ;计算 y=10×x

          MOV B,60H       ;X为(61H,60H)

          MUL AB

          MOV 67H,A       ;积为(69H,68,67H)

          MOV 68H,B

          MOV A,#0AH

          MOV B,61H

          MUL AB

          ADDC A,68H

          MOV 68H,A

          MOV A,B

          ADDC A,#00H

          MOV 69H,A

          MOV 43H,#00H    ;计算(10×x)/m

          MOV 42H,#00H    ;被除数(69H,68H,67H) 除数(37H)

          MOV 62H,#00H    ;商(43H,42H) 余数(62H)

    DIVD: CLR C

          MOV A,67H

          SUBB A,37H

          MOV B,A

          JC QUIT1       ;不够减的话转至 QUIT1

          MOV 67H,A

          AJMP G         ;够减的话转至 G

   QUIT1: MOV A,68H

          JZ QUIT3  ;最低位不够减且第二位为0,向最高位借位转QUIT3

   QUIT2: MOV 67H,B ;最低位不够减但第二位不为0时

          MOV A,68H

          SUBB A,#00H

          MOV 68H,A

          AJMP G     ;此部分处理完毕后转至 G

   QUIT3: MOV A,69H

          JZ SUBB1   ;借位但最高位和第二位都为0时除法转换结束

          SUBB A,#00H

          MOV 69H,A

          MOV 68H,#0FFH

          MOV A,#0FFH

          SUBB A,37H

          ADD A,67H

          INC A

          MOV 67H,A

       G: MOV 62H,67H        ;62H放余数

          MOV A,42H

          ADD A,#01H

          MOV 42H,A

          MOV A,43H

          ADDC A,#00H

          MOV 43H,A

          AJMP DIVD

   SUBB1: MOV A,62H          ;处理余数

          CJNE A,#5DH,SUBB2  ;不等转SUBB2,进行下一步判别

          JC SUBB3           ;相等(5入),跳转至SUBB3

   SUBB2: JC SUBB4           ;(62H)<5DH,(4舍)并跳转至SUBB4

   SUBB3: CLR C

          MOV A,42H

          ADD A,#01H

          MOV 42H,A

          MOV A,43H

          ADDC A,#00H

          MOV 43H,A

   SUBB4: CLR C              ;计算 (10×x/m) - n

          MOV A,42H

          SUBB A,38H

          MOV 42H,A

          MOV A,43H

          SUBB A,#00H

          MOV 43H,A

    HTOB: MOV 56H,#00H        ;16进制(43H,42H)→BCD码

          MOV 57H,#00H        ;BCD码:5AH(千位)-56H(小数位)

          MOV 58H,#00H        ;测量值范围(00.0-99.9)

          MOV 59H,#00H        ;实际存放(58H,57H,56H)

          MOV 5AH,#00H

   HWWEI: CLR C               ;处理千位

          MOV A,42H

          SUBB A,#10H

          MOV B,A

          MOV A,43H           ;2710H=10000B(含小数位故为五位数)

          SUBB A,#27H

          JC HQWEI

          INC 5AH

          MOV 43H,A

          MOV 42H,B

          AJMP HWWEI

   HQWEI: CLR C                ;处理百位

          MOV A,42H

          SUBB A,#0E8H

          MOV B,A

          MOV A,43H            ;03E8H=1000B

          SUBB A,#03H

          JC HBWEI

          INC 59H

          MOV 43H,A

          MOV 42H,B

          AJMP HQWEI

   HBWEI: CLR C                 ;处理十位

          MOV A,42H

          SUBB A,#64H

          MOV B,A

          MOV A,43H             ;64H=100B

          SUBB A,#00H

          JC HSWEI

          INC 58H

          MOV 43H,A

          MOV 42H,B

          AJMP HBWEI

   HSWEI: CLR C                  ;处理个位

          MOV A,42H

          SUBB A,#0AH

          JC HGWEI               ;0AH=10B

          INC 57H

          MOV 42H,A

          AJMP HSWEI

   HGWEI: MOV 56H,42H            ;处理小数位

     QU2: RET

 CHAJIAN: LCALL LED           ;键盘子程序

          MOV DPTR,#6003H

          MOV A,#81H          ;8255(2)PC口高位输出、低位输入

          MOVX @DPTR,A

          MOV A,#00H          ;PC口高位送全0

          MOV DPTR,#6002H

          MOVX @DPTR,A

          MOVX A,@DPTR         ;PC口低位读入

          ANL A,#0FH

          CJNE A,#0FH,GOON1

          MOV R2,#0FFH

          AJMP FH              ;无键按下返回

   GOON1: MOV 30H,A            ;存入30H

          MOV DPTR,#6003H

          MOV A,#88H           ;8255(2)PC口高位输入、低位输出

          MOVX @DPTR,A

          MOV A,#00H           ;PC口低位送全0

          MOV DPTR,#6002H

          MOVX @DPTR,A

          MOVX A,@DPTR         ;PC口高位读入

          ANL A,#0F0H

          ADD A,30H            ;合成后放入30H

          MOV 30H,A

          LCALL DELAY20        ;延时去抖动

     ZAI: MOV DPTR,#6003H

          MOV A,#81H           ;再读一次

          MOVX @DPTR,A

          MOV DPTR,#6002H

          MOV A,#00H

          MOVX @DPTR,A

          MOVX A,@DPTR

          ANL A,#0FH

          MOV R1,A

          MOV DPTR,#6003H

          MOV A,#88H

          MOVX @DPTR,A

          MOV DPTR,#6002H

          MOV A,#00H

          MOVX @DPTR,A

          MOVX A,@DPTR

          ANL A,#0F0H

          ADD A,R1                 ;合成后放入A

          CJNE A,30H,CHAJIAN       ;进行比较,如不同重读键盘

          MOV R1,#10H              ;键值表的长度

          MOV R2,#00H              ;R2为键盘标称值

  SEARCH: MOV DPTR,#KEYTABLE       ;键值表的首地址

          CLR C

          MOV A,R2

          MOVC A,@A+DPTR

          SUBB A,30H                ;进行比较

          JZ FIND                   ;查到了

          INC R2

          DJNZ R1,SEARCH

          MOV R2,#0FFH

          AJMP FH                   ;查完整个表没查到返回

    FIND: LCALL LED

          MOV DPTR,#6003H

          MOV A,#81H                ;等待按键松下

          MOVX @DPTR,A

          MOV A,#00H

          MOV DPTR,#6002H

          MOVX @DPTR,A

          MOVX A,@DPTR

          ANL A,#0FH

          CJNE A,#0FH,FIND          ;无键按下时表明按键释放

    FH:   RET

 CONTROL: CLR C     ;控制子程序

          MOV A,4FH ;设定值(4FH,4EH,4DH)测量值(58H,57H,56H)

          CJNE A,58H,CH11

          MOV A,#0AH ;(4FH=58H时)

          MOV B,A    ;设定值个位小数位(4EH,4DH)→16进制(5DH)

          MOV A,4EH

          MUL AB

          ADD A,4DH

          MOV 5DH,A

          MOV A,#0AH ;测量值个位小数位(57H,56H)→16进制(5EH)

          MOV B,A

          MOV A,57H

          MUL AB

          ADD A,56H

          MOV 5EH,A

          CJNE A,5DH,CH1

          AJMP CHU1  ;(设定值=测量值)关

  CH1:    CLR C

          MOV A,5DH

          SUBB A,5EH ;设定值个位小数位(5DH)-测量值个位小数位(5EH)

          JC CHU1     ;(设定值<测量值)关

          SUBB A,#3CH ;(3CH=60D)

          JNC CHU2    ;(设定值-测量值)>=6.0℃开

          CLR C

          MOV A,5DH

          SUBB A,5EH

          SUBB A,#28H ;(28H=40D)

          JNC CH2     ;(设定值-测量值)在6℃-4℃

          CLR C

          MOV A,5DH

          SUBB A,5EH

          SUBB A,#14H ;(14H=20D)

          JNC CH3     ;(设定值-测量值)在4℃-2℃

          AJMP CH6    ;(设定值-测量值)在2℃-0℃

   CH2:   MOV A,2EH

          CJNE A,#06H,CHU2  ;(设定值-测量值)在6℃-4℃时

          AJMP CHU1         ;加热开/关时间比5:1

   CH3:   MOV A,2EH

          CJNE A,#01H,CH4 ;(设定值-测量值)在4℃-2℃时

          AJMP CHU2

   CH4:   CJNE A,#03H,CH5 ;加热开/关时间比3:3

          AJMP CHU2

   CH5:   CJNE A,#05H,CHU1

          AJMP CHU2

   CH6:   MOV A,2EH

          CJNE A,#01H,CHU1  ;(设定值-测量值)在2℃-0℃时

          AJMP CHU2         ;加热开/关时间比1:5

   CH11:  JC CHU1      ;(4FH<>58H时)

          SUBB A,58H   ;(设定值<测量值)关

          SUBB A,#02H

          JNC CHU2     ;(设定值十位-测量值十位)>=2开

          MOV A,#0AH ;(设定值十位-测量值十位)=1时

          MOV B,A    ;设定值个位小数位(4EH,4DH)→16进制(5DH)

          MOV A,4EH

          MUL AB

          ADD A,4DH

          MOV 5DH,A

          MOV A,#0AH ;测量值个位小数位(57H,56H)→16进制(5EH)

          MOV B,A

          MOV A,57H

          MUL AB

          ADD A,56H

          MOV 5EH,A

          CLR C

          MOV A,5DH

          ADD A,#64H ;(64H=100D)

          MOV 5DH,A

          SUBB A,5EH

          SUBB A,#3CH ;(3CH=60D)

          JNC CHU2    ;(设定值-测量值)>=6.0℃开

          CLR C

          MOV A,5DH

          SUBB A,5EH

          SUBB A,#28H ;(28H=40D)

          JNC CH2     ;(设定值-测量值)在6℃-4℃

          CLR C

          MOV A,5DH

          SUBB A,5EH

          SUBB A,#14H ;(14H=20D)

          JNC CH3     ;(设定值-测量值)在4℃-2℃

          AJMP CH6    ;(设定值-测量值)在2℃-0℃

    CHU1: MOV A,#00H

          MOV DPTR,#0E002H

          MOVX @DPTR,A      ;8255(1)PC0关掉加热开关

          AJMP CHU3

    CHU2: MOV A,#01H

          MOV DPTR,#0E002H

          MOVX @DPTR,A      ;8255(1)PC0打开加热开关

    CHU3: RET

          ORG 1000H

   INTER: CLR EA

          PUSH PSW             ;现场保护

          PUSH ACC

          SETB RS1

          CLR RS0             ;二区工作寄存器

     D5:  MOV DPTR,#0E000H

          MOVX A,@DPTR         ;8255(1)PA7-PA4全为低为万位

          MOV R3,A

          ANL A,#0FH

          JNZ D5

          MOV A,R3

          ANL A,#0F0H

          SWAP A

          MOV 54H,A               ;放入地址54H(万位)

     D4:  MOV DPTR,#0E000H

          MOVX A,@DPTR

          MOV R3,A

          ANL A,#08H

          JZ D4

          MOV A,R3

          ANL A,#0F0H

          SWAP A

          MOV 53H,A               ;放入地址53H(千位)

     D3:  MOV DPTR,#0E000H

          MOVX A,@DPTR

          MOV R3,A

          ANL A,#04H

          JZ D3

          MOV A,R3

          ANL A,#0F0H

          SWAP A

          MOV 52H,A                ;52H(百位)

     D2:  MOV DPTR,#0E000H

          MOVX A,@DPTR

          MOV R3,A

          ANL A,#02H

          JZ D2

          MOV A,R3

          ANL A,#0F0H

          SWAP A

          MOV 51H,A                ;51H(十位)

     D1:  MOV DPTR,#0E000H

          MOVX A,@DPTR

          MOV R3,A

          ANL A,#01H

          JZ D1

          MOV A,R3

          ANL A,#0F0H

          SWAP A

          MOV 50H,A                 ;50H(个位)

  RETURN: ACALL BTOH

          MOV A,2CH

          INC A

          MOV 2CH,A

          MOV R0,2DH

          MOV @R0,5BH

          INC R0

          MOV @R0,5CH

          MOV A,2DH

          INC A

          INC A

          MOV 2DH,A

          POP ACC               ;恢复现场

          POP PSW

          SETB EA               ;开中断

          RETI

    BTOH: MOV 5BH,50H    ;BCD码(54H-50H)→16进制(5CH,5BH)

          MOV 5CH,#00H   ;个位

   BSWEI: MOV A,51H        ;十位

          JZ BBWEI

          MOV R7,51H

      SS: MOV A,5BH

          ADD A,#0AH       ;0AH=10B

          MOV 5BH,A

          MOV A,5CH

          ADDC A,#00H

          MOV 5CH,A

          DJNZ R7,SS

   BBWEI: MOV A,52H        ;百位

          JZ BQWEI

          MOV R7,52H

      BB: MOV A,5BH

          ADD A,#64H       ;64H=100B

          MOV 5BH,A

          MOV A,5CH

          ADDC A,#00H

          MOV 5CH,A

          DJNZ R7,BB

   BQWEI: MOV A,53H         ;千位

          JZ BWWEI

          MOV R7,53H

      QQ: MOV A,5BH

          ADD A,#0E8H

          MOV 5BH,A

          MOV A,5CH          ;03E8H=1000B

          ADDC A,#03H

          MOV 5CH,A

          DJNZ R7,QQ

   BWWEI: MOV A,54H          ;万位

          JZ QU1

          MOV R7,54H

      WW: MOV A,5BH

          ADD A,#10H

          MOV 5BH,A

          MOV A,5CH          ;2710H=10000B

          ADDC A,#27H

          MOV 5CH,A

          DJNZ R7,WW

     QU1: RET

 DELAY20: MOV R7,#0AAH

      AA: MOV R6,#0FFH

      DD: DJNZ R6,DD

          DJNZ R7,AA

          RET

   DELAY: MOV R7,#02H

  DELAY0: MOV R6,#0FFH

  DELAY1: DJNZ R6,DELAY1

          DJNZ R7,DELAY0

          RET

     SEG: DB 40H,79H,24H,30H,19H,12H,02H,78H

          DB 00H,10H,08H,03H,46H,21H,06H,0EH

          DB 7FH,3FH,09H                            ;段码

   TABLE: DB 0D3H,09H,0CAH,09H,0CCH,0CH,0C3H,1AH

          DB 0C6H,13H,0C5H,16H,0C3H,1EH,0C4H,19H

          DB 0BAH,43H,0A2H,0D6H                     ;参数m,n

KEYTABLE: DB 7EH,0BEH,0BDH,0BBH,0DEH,0DDH,0DBH,0EEH

          DB 0EDH,0EBH,0E7H,0D7H,0B7H,77H,7BH,7DH   ;键值表

          END

第十一章  自主设计型实验

实验二十二  简易数字电压表设计
一、实验目的：

了解直流电压的测量方法，掌握直流电压测量系统中软硬件的设计。

二、实验内容：

利用电压衰减器、 A/D转换器（ADC0809）、8031 单片机测量直流电压；

测量结果在LED显示器上显示；

测量、显示范围为00.0V～12.0V。

三、硬件连接框图见图11-1。

相关电路原理图见第一章图1-4、图1-6、图1-9。

四、实验步骤

1、 硬件连接

1） 将+12V电压接入电位器WR1的一端，0V（GND）接入另一端。中间抽头输出0V~12V，此电压接入电压衰减器输入端Vin2；
2) 调节WR4使电压衰减器衰减系数为1/3（如调节WR1使Vin2输入9V，调

节WR4使Vout2输出3V）；

3） 电压衰减器输出端Vout2接ADC0809输入IN0；
4） ADC0809读、写信号控制端 /RD、/WR接8031的 /RD、/WR；

ADC0809时钟信号CLOCK接1/2ALE；

ADC0809片选信号 /CS接 /Y3~/Y7（任选其一）；

ADC0809输出D7~D0接8031的P0.7~P0.0；

5） 8255(2) 片选信号 /CS接 /Y3~/Y7（任选其一）；

6） 连接相关硬件模块的电源短路块。

2、 软件设计：

    A/D数据采集后，将数据×3。转换成BCD码后送去显示，显示时加小数点。

1） 8255（2）初始化（PA口、PB口输出，扩展LED显示器）；

2） 8255（2）PA口PA7控制小数点，PA6~PA0控制字形（反向驱动），PB口PB5~PB0控制字位（反向驱动）。LED显示器为共阴极；

3） 数据采集采用延时方式，驱动A/D转换后，延时150μS左右，读取数据；

五、设计程序，执行程序，检查结果。

实验二十三  简易温度测量系统设计
一、实验目的：

了解温度测量的方法，掌握温度测量系统中软硬件的设计。

二、实验内容：

利用Pt100热电阻、测量电桥、测量放大器、 A/D转换器（ADC0809）、8031 

单片机测量水温；

在LED显示器上显示测量水温；

测量、显示水温范围为00.0˚C～99.9˚C；

在程序设计调试过程中，利用电阻箱模拟Pt100热电阻的阻值变化。
三、硬件连接框图见图11-2。

相关电路原理图见第一章图1-2、图1-3、图1-6、图1-9。

四、实验步骤

1、 硬件连接

1） 电阻箱接入测量电桥输入端Rx；

2） 电阻箱阻值设定为100.0Ω（Pt100热电阻在水温为0˚C时的阻值），

调节电位器WR3（调零电位器）,使测量电桥输出Vout11、Vout12电压为0V；

3） 测量电桥输出Vout11、Vout12接入测量放大器输入端Vin41、Vin42，

调节电位器WR5（调零电位器），使测量放大器输出Vout4电压为0V；

4） 将测量放大器输出Vout4接入ADC0809转换电路输入端IN0；

5） ADC0809读、写信号控制端 /RD、/WR接8031的 /RD、/WR；

ADC0809时钟信号CLOCK接1/2ALE；

ADC0809片选信号 /CS接 /Y3~/Y7（任选其一）；

ADC0809输出D7~D0接8031的P0.7~P0.0；

5） 8255(2) 片选信号 /CS接 /Y3~/Y7（任选其一）；

6） 连接相关硬件模块的电源短路块。

2、 软件设计说明：

A/D转换后将数据保存；

六次A/D转换后进行数字滤波，得到A/D转换有效值；

根据A/D转换有效值进行非线性校正及标度变换，得到测量温度值（十六进制数）；

再将此十六进制温度值变换成BCD码（000~999），送去显示。显示时个位数加小数点（00.0˚C ~99.9˚C）。

1） 设计ADC0809转换程序

2） 确定零点及满量程：

执行A/D转换程序；

调整电阻箱的阻值为100.0Ω，调节电位器WR5（调零电位器）使显示值为10H左右；

调整电阻箱的阻值为138.1Ω，调节电位器WR4（调增益电位器）使显示值为F0H左右；

重复调整，达到以上要求。10H及F0H为判断欠量程、过量程的阀值。

3） 建立表格：（按下列表格调整电阻箱的阻值，记录显示数据。）

	温度（℃）
	  0．0
	10．0
	20．0
	30．0
	40．0
	50．0

	阻值（Ω）
	 100.0
	103.9
	107.8
	111.7
	115.5
	119.4

	A/D数值
	 
	
	
	
	
	

	十进制
	
	
	
	
	
	

	
	
	
	
	
	
	

	温度（℃）
	60．0
	70．0
	80．0
	90．0
	99．9
	

	阻值（Ω）
	123.2
	127.1
	130.9
	134.7
	138.1
	

	A/D数值
	
	
	
	
	
	

	十进制
	
	
	
	
	
	


4） 保存六次A/D转换的数值

5） 数字滤波程序设计：

当读入A/D转换后数值的次数达到6次时，执行数字滤波程序（去极值求

平均值法即去掉最高值及最低值，然后求平均值）；

求得A/D转换的有效值并保存。

6） 设计非线性校正及标度变换程序：
1 将上表按温度均匀分成十段，以温度值和A/D采集数据为函数关系建立十

段线性插值公式。

    设直线方程y=ax+b。y是温度值，x是A/D采集数据。

    从表中可以得出每段的（x1,y1）和(x2,y2)。

    求解方程 y1 = ax1+b           则  a = (y2-y1) / (x2-x1)

             y2 = ax2+b               b = (x1y2-x2y1) / (x1-x2)          
②  根据A/D转换的有效值的大小，选择插值公式，求出被测温度值y。

7） 16进制数→BCD码转换程序设计：
根据插值公式求得的y值为16进制数，将其转换成BCD码（范围为000～

999），再将其送入显示缓冲区。
小数点在确定个位数的字形码时给出,显示范围为00.0˚C～99.9˚C。
8） 当A/D转换的有效值超出欠量程、过量程的阀值时，给出欠量程或过量程

标志（欠量程显示全“－”，过量程显示全“H” ）。

9） 根据以上说明设计程序（在ADC0809转换程序上修改），将温度值在右边三位LED显示器上显示。

3、 执行程序：

    使用电阻箱模拟Pt100的阻值变化，记录测量结果；

使用Pt100替换电阻箱，记录测量结果。

实验二十四  简易温度测量控制系统设计
一、实验目的：

了解温度测量控制的方法，掌握温度测量控制系统中软硬件的设计。

二、实验内容：

利用Pt100热电阻、测量电桥、测量放大器、 A/D转换器（ADC0809）、8031 

单片机测量水温；

利用固态继电器、电热杯控制加热水温；

利用键盘设定温度并控制系统工作；

LED显示器左边三位显示设定水温，右边三位显示测量水温；

测量、控制、显示水温范围为00.0˚C～99.9˚C；

在程序设计调试过程中，利用电阻箱模拟Pt100热电阻的阻值变化。
三、硬件连接框图见图11-3。

相关电路原理图见第一章图1-2、图1-3、图1-6、图1-8、图1-9、图1-11。


[image: image24.emf]电热杯

固态

继电器

8255(1)

PC口

Pt100

热电阻

测量

电桥

测量

放电路

A/D

8051

单片机

键盘

显示器

8255(2)

PA口

PB口


图11-3

四、实验步骤：

1、 硬件连接

1） 电阻箱接入测量电桥输入端Rx；

2） 电阻箱阻值设定为100.0Ω（Pt100热电阻在水温为0˚C时的阻值），

调节电位器WR3（调零电位器）,使测量电桥输出Vout11、Vout12电压为0V；

3） 测量电桥输出Vout11、Vout12接入测量放大器输入端Vin41、Vin42，

调节电位器WR5（调零电位器），使测量放大器输出Vout4电压为0V；

4） 将测量放大器输出Vout4接入ADC0809转换电路输入端IN0；

5） ADC0809读、写信号控制端 /RD、/WR接8031的 /RD、/WR；

ADC0809时钟信号CLOCK接1/2ALE；

ADC0809片选信号 /CS接 /Y3~/Y7（任选其一）；

ADC0809输出D7~D0接8031的P0.7~P0.0；

6） 8255(2) 片选信号 /CS接 /Y3~/Y7（任选其一）；

7） 控制通道：

8255（1）PC口的PC0接CONin；

CONout(+/-) 接固态继电器控制信号输入端(+/-) ；

固态继电器控制信号输出端接电热杯；

固态继电器电源端接交流220V；

电热杯内必须加水！    

8） 连接相关硬件模块的电源短路块。

2、 软件设计说明：

利用键盘设定温度并控制系统工作；

LED显示器左边三位显示设定温度，右边三位显示测量温度；

A/D转换后将数据保存；

六次A/D转换后进行数字滤波，得到A/D转换有效值；

根据A/D转换有效值进行非线性校正及标度变换，得到测量温度值（十六进制数）；

再将此十六进制温度值变换成BCD码（000~999），送去显示，显示时个位数加小数点（00.0˚C ~99.9˚C）；

    将设定值与测量值进行比较，确定控制通道的工作。

1） 设计ADC0809转换程序

2） 确定零点及满量程：

  执行A/D转换程序；
  调整电阻箱的阻值为100.0Ω，调节电位器WR5（调零电位器）使显示值为10H左右；

  调整电阻箱的阻值为138.1Ω，调节电位器WR4（调增益电位器）使显示值为F0H左右；

  重复调整，达到以上要求；

  10H及F0H为判断欠量程、过量程的阀值。

3） 建立表格：（按下列表格调整电阻箱的阻值，记录显示数据。）

	温度（℃）
	  0．0
	10．0
	20．0
	30．0
	40．0
	50．0

	阻值（Ω）
	 100.0
	103.9
	107.8
	111.7
	115.5
	119.4

	A/D数值
	 
	
	
	
	
	

	十进制
	
	
	
	
	
	

	
	
	
	
	
	
	

	温度（℃）
	60．0
	70．0
	80．0
	90．0
	99．9
	

	阻值（Ω）
	123.2
	127.1
	130.9
	134.7
	138.1
	

	A/D数值
	
	
	
	
	
	

	十进制
	
	
	
	
	
	


4） 读入六次A/D转换数值，将其保存。

5） 数字滤波程序设计：

当读入A/D转换后的数值次数达到6次时，执行数字滤波程序（去极值求平均值法即去掉最高值及最低值，然后求平均值）；

求得A/D转换的有效值并保存。

6） 设计非线性校正及标度变换程序：

①  建立插值公式：

将上表按温度均匀分成十段，以温度值和A/D采集数据为函数关系建立十段线性插值公式。

    设直线方程y=ax+b。y是温度值，x是A/D采集数据。

    从表中可以得出每段的（x1,y1）和(x2,y2)。

    求解方程 y1=ax1+b           则 a=(y2-y1)/(x2-x1)

             y2=ax2+b                 b= (x1y2-x2y1) / (x1-x2)
②  根据A/D转换的有效值的大小，选择插值公式，求出被测温度值y。
7） 16进制数→BCD码转换程序设计：
根据插值公式求得的y值为16进制数，将其转换成BCD码（范围为000～

999）并保存。
8） 键盘设定温度并控制系统工作：

    数字键为0~9。

A键为工作键，B键为暂停（设定）键。

9） 根据以上要求设计程序（修改ADC0809转换程序）；

测量温度值在右边三位LED显示器上显示；

设定温度值在左边三位LED显示器上显示；

显示范围均为00.0˚C～99.9˚C。小数点在确定个位数的字形码时给出；

    当A/D转换的有效值超出欠量程、过量程的阀值时，给出欠量程或过量程

标志（欠量程显示全“－”，过量程显示全“H” ）。

10）温度控制方式：

    采用脉宽调制方式。

当（设定键─测量值）≧ 6˚C 时，全加热；

当（设定键─测量值）在6˚C ~ 4˚C 时，加热通/断时间比5:1 ；         

当（设定键─测量值）在4˚C ~ 2˚C 时，加热通/断时间比1:1 ；

当（设定键─测量值）在2˚C ~ 0˚C 时，加热通/断时间比1:5 ；

当（设定键─测量值）≦0˚C 时，全关断。

3、 根据以上要求设计（修改）程序；

执行程序；

使用电阻箱模拟Pt100的阻值变化，记录测量控制结果；

使用Pt100替换电阻箱，将Pt100插入电热杯中，记录测量控制结果。

附录：  实验仪面板示意图

